IREM DE GRENOBLE

INITIATION À MAPLE V
fiches de formation

GROUPE CALCUL FORMEL

Philippe BIZARD
Jean-Pierre DOURIS
Jean-Louis LANGON
Jacques MARTINIE
Paul PERRET
Joël PINCHINAT

INITIATION À MAPLE V
Fiches de formation

document de travail du 15 décembre 1996

ISBN2 - 90-38-15-35-6
PRIX : 30F

1e édition 1997

Sommaire

Fiches

Annexe

Découverte de Maple V

Un logiciel ouvert

Dès le lancement de Maple V, l'utilisateur a la possibilité de créer.
Tous ses ordres sont immédiatement interprétés et exécutés.

Importance du ;

L'entrée « > 2+2 ; » provoque l'affichage du résultat 4.
Le ; est indispensable. C'est lui qui déclenche le calcul et l'affichage du résultat.

Affectation

L'affectation permet de créer une variable ou de la modifier et donc de mémoriser une valeur ou plus généralement un objet.
Par exemple :

> a := 2+2 ;	permet de donner à la variable a le contenu 4.
> tva := prix * 0.206 ;	place dans la variable tva la taxe relative à la valeur contenue dans la variable prix

Les chaînes 'a', 'tva' et 'prix' sont des noms de variables.

Libération d'une variable

Il peut être gênant qu'une variable que l'on veut utiliser ait déjà été affectée.
Il est possible de libérer une variable en lui donnant son nom comme contenu.
Par exemple : tva := 'tva' .

Les fiches qui suivent regroupent, par thèmes, quelques unes des commandes de Maple V. Notre objectif est de faire découvrir quelques unes des principales possibilités du logiciel.
Nous souhaitons convaincre nos collègues scientifiques que ce logiciel leur est immédiatement accessible.

Quelques connaissances de l'environnement Windows facilitent le travail.
Pour les débutants, l'annexe "Maple V dans l'environnement Windows" devrait suffire.

Si l'on charge un fichier Maple V, une fiche par exemple, on n'est pas obligé de l'exécuter de façon linéaire.

On peut, n'importe où dans le texte, créer une nouvelle zone d'entrée, pour essayer, suivant sa propre inspiration, d'autres exemples, variés et nombreux, d'une même commande.
Le texte d'origine donne un exemple de la syntaxe d'une commande.
Cet exemple est très souvent suffisant pour comprendre, tout à la fois, la signification d'une commande et son utilisation.

Vous trouverez en annexe un résumé des principales commandes utilisées dans ces fiches.

Une ambiance fonctionnelle.

Maple V manipule essentiellement des objets tels que nombres, séquences, listes...
Les commandes sont essentiellement fonctionnelles : elles rendent donc des objets que l'utilisateur peut, à son gré, afficher, mémoriser ou utiliser dans d'autres ordres ou séquences d'ordres.

L'objet x -> x^2 est bien la fonction connue en mathématique.
Cet objet peut être affecté à une variable, on peut dire aussi nommé.
Après l'exécution de la commande : > f := x -> x^2 ;
f est bien une fonction mathématique, f(x) l'image de x... f(2), f(t) sont alors connus.
> plot(f) ; retourne le graphe de f.

L'objet f est alors utilisable : par exemple f@f est la composée de f par f et D(f) est la fonction dérivée de f.

Maple V dans l'environnement Windows

Contraintes matérielles

Nous avons travaillé essentiellement sur des ordinateurs PC.
Le poste utilisé doit être suffisamment performant, à savoir, équipé d'un microprocesseur i80486 ou pentium, et disposer d'une mémoire vive de 8Mo ou mieux encore 16 Mo.
Windows (version 3.1x ou postérieure) doit être installé ainsi qu'une version de Maple V prévue pour Windows (R3, R4, ou ultérieure)
Nous n'avons pas beaucoup d'expérience sur d'autres plates-formes.

Mise en route

1) Lancer Windows

2) Cliquer 2 fois l'icône de Maple V

La célèbre feuille d'érable...

Chargement et sauvegarde

Pour charger un fichier Maple V

Utiliser la commande 'Ouvrir', dans le menu déroulant 'Fichier'.
On peut accéder au fichier à ouvrir en sélectionnant, à l'aide de la souris, le lecteur de disquettes désiré et/ou le répertoire adéquat.
Attention ! Les seuls fichiers utilisables doivent avoir l'extension '.ms' en version R3 ou '.msw' en version R4.
On peut également taper le nom complet du fichier : par exemple 'A:\fiches\arith.mws'.

Pour sauvegarder une fiche de travail

Utiliser la commande 'Save', dans le menu déroulant 'File'.
Donner le nom que vous voulez attribuer au fichier.
Assurer vous que le nom du fichier comporte l'extension '.ms' ou '.mws'
Le fichier sera alors sauvegardé, dans le répertoire actif, et sera accessible grâce à son nom.
Une fois que la première sauvegarde est faite, les mises à jour seront réalisées plus simplement toujours grâce à la commande 'Save' , dans le menu déroulant 'File' ou par la commande 'Save as'.

L'éditeur plein écran

Entrée d'une commande

En général les commandes - Maple V sont écrites dans une zone réservée aux **Entrées (Input zone)**
Une telle zone est reconnaissable par le symbole d'invite : '>'.
On peut matérialiser les différentes zones d'entrée, à l'aide de lignes séparatrices, en cliquant l'icône approprié.
Une zone d'entrée peut être constituée de plusieurs lignes contiguës.

Une commande est exécutée par l'appui sur la touche 'Entrée', à condition que la zone d'entrée soit bien terminée par le caractère ' ;' ou par le caractère ' :'

Une commande terminée par ';' est exécutée et provoque l'affichage dans une zone de sortie (Output zone).
Exemple : > a :=2 + 2 ;
C'est donc le point-virgule que nous utiliserons le plus souvent.

Une commande terminée par ':' est exécutée sans provoquer d'afichage.
Autrement dit on ne voit rien, mais l'objet est bien créé.
Exemple : > a :=2 + 2 :

Les **commandes graphiques** peuvent si on le souhaite provoquer l'ouverture de fenêtres graphiques où les tracés sont réalisés.

Les possibilités d'édition

Ce sont celles de Windows (couper, copier, coller).

L'impression

Le document actif peut être imprimé à tout moment en utilisant l'option 'Print' du menu 'File' ou l'icône correspondant, pour peu que le poste soit en liaison active avec une imprimante...
Les fenêtres graphiques peuvent aussi être imprimées.

Dangers

Puisque l'on peut se déplacer à l'envi dans un texte écrit, on peut envoyer des commandes, qui ne se suivent pas forcément dans le texte.
On peut ainsi, parfois, ne plus bien contrôler certaines variables, et faire par conséquent quelques confusions.

Aide

Maple V fournit une aide en ligne, accessible au menu par le symbole ?
Pour obtenir une aide sur une commande particulière et connue, on gagnera du temps en tapant, dans une zone d'entrée, sur une ligne vierge, le point d'interrogation suivi du nom de la commande, suivi de 'Entrée' ou en se plaçant sur un mot et en appuyant sur F1.
Exemple : > ?evalf

Calculs numériques

Groupe calcul formel - IREM de Grenoble

Fiche n° 1

```
> restart;
> 2+3 ; 5*2 ;
```

$$5$$
$$10$$

```
> 2^3 ; 3*2^5+4^2 ;
```

$$8$$
$$112$$

```
> 5/8+2/16 ; (3/4)/(9/16) ;
```

$$\frac{3}{4}$$
$$\frac{4}{3}$$

```
> for i from 1 to 9 do seq(i*j, j = 1..9) od ;
```

1, 2, 3, 4, 5, 6, 7, 8, 9
2, 4, 6, 8, 10, 12, 14, 16, 18
3, 6, 9, 12, 15, 18, 21, 24, 27
4, 8, 12, 16, 20, 24, 28, 32, 36
5, 10, 15, 20, 25, 30, 35, 40, 45
6, 12, 18, 24, 30, 36, 42, 48, 54
7, 14, 21, 28, 35, 42, 49, 56, 63
8, 16, 24, 32, 40, 48, 56, 64, 72
9, 18, 27, 36, 45, 54, 63, 72, 81

```
> 1/(1+sqrt(2)) ;
```

$$\frac{1}{1+\sqrt{2}}$$

```
> (1/(1+sqrt(2)))^3;
```

$$\frac{1}{(1+\sqrt{2})^3}$$

```
> expand((1 + sqrt(3))^3) ;
```

$$10+6\sqrt{3}$$

Calculs approchés

```
> evalf(Pi,100) ;
```

3.141592653589793238462643383279502884197169399375105820974944592307816406286\
    208998628034825342117068

```
> evalf(2^(1/2),20) ;
```

$$1.41421356237309504 88$$

```
> n := (1+2^(1/2))^1994 ;
```

$$n := (1+\sqrt{2})^{1994}$$

```
> evalf(n) ;
```

$$.1797696736\ 10^{764}$$

```
> evalf(n-floor(n), 1000) ;
```

999999999999999999999999999999999999999999999999999999999999999999999999\
    999999999999999999999999999999999999999999999999999999999999999999999999\
    999999999999999999999999999999999999999999999999999999999999999999999999\
    99999999690

le nombre de chiffres est défini par la variable Digits

```
> Digits ;
```

10

```
> Digits := 100 ; evalf(sqrt(3)) ;
```

Digits := 100

1.732050807568877293527446341505872366942805253810380628055806979451933016908\
    8000370811461867572485756

```
> fsolve(cos(x)=x) ;
```

.7390851332151606416553120876738734040134117589007574649656806357732846548835\
    47594599376106931766531 9

```
> Digits := 10 ; fsolve(x^7=1, x, complex) ;
```

Digits := 10

$-.9009688679 - .4338837391\,I, -.9009688679 + .4338837391\,I,$
$\quad -.2225209340 - .9749279122\,I, -.2225209340 + .9749279122\,I,$
$\quad .6234898019 - .7818314825\,I, .6234898019 + .7818314825\,I, 1.$

Nombres rationnels

```
> evalf(1/17, 17) ; evalf(1/17, 33) ;evalf(1/17, 49) ;
```

.05882352941176470 6

.058823529411764705882352941176470 6

.058823529411764705882352941176470588235294117647 06

```
> evalf(8/21, 7) ; evalf(8/21, 13) ; evalf(8/21, 19) ;
```

.3809524

.3809523809524

.3809523809523809524

```
> evalf(5/17,17) ; evalf(5/17,33) ; evalf(5/17,49) ;
```

.29411764705882353

.294117647058823529411764705882353

29411764705882352941176470588235294117647058823 53

```
>
```

Factorielle

Groupe calcul formel - IREM de Grenoble

Fiche n° 2

```
> restart;
> 10! ;
```

3628800

```
> 100! ;
```

93326215443944152681699238856266700490715968264381621468592963895217599993229915608941463976156518286253697920827223758251185210916864000000000000000000000000

Quels sont les facteurs premiers de 100!

```
> ifactor(100!) ;
```

$(2)^{97} (3)^{48} (5)^{24} (7)^{16} (11)^{9} (13)^{7} (17)^{5} (19)^{5} (23)^{4} (29)^{3} (31)^{3} (37)^{2} (41)^{2} (43)^{2} (47)^{2} (53)(59)(61)(67)(71)(73)(79)(83)(89)(97)$

Combien de zéros y-a-t'il à la fin de 100!

```
> 100!/10^24 ;
```

9332621544394415268169923885626670049071596826438162146859296389521759999322991560894146397615651828625369792082722375825118521091686 4

```
> t := time() : ifactor(1024!) ;  t := time() - t ;
```

$(2)^{1023} (3)^{508} (5)^{253} (7)^{168} (11)^{101} (13)^{84} (17)^{63} (19)^{55} (23)^{45} (29)^{36} (31)^{34} (37)^{27} (41)^{24} (43)^{23} (47)^{21} (53)^{19} (59)^{17} (61)^{16} (67)^{15} (71)^{14} (73)^{14} (79)^{12} (83)^{12} (89)^{11} (97)^{10} (101)^{10} (103)^{9} (107)^{9} (109)^{9} (113)^{9} (127)^{8} (131)^{7} (137)^{7} (139)^{7} (149)^{6} (151)^{6} (157)^{6} (163)^{6} (167)^{6} (173)^{5} (179)^{5} (181)^{5} (191)^{5} (193)^{5} (197)^{5} (199)^{5} (211)^{4} (223)^{4} (227)^{4} (229)^{4} (233)^{4} (239)^{4} (241)^{4} (251)^{4} (257)^{3} (263)^{3} (269)^{3} (271)^{3} (277)^{3} (281)^{3} (283)^{3} (293)^{3} (307)^{3} (311)^{3} (313)^{3} (317)^{3} (331)^{3} (337)^{3} (347)^{2} (349)^{2} (353)^{2} (359)^{2} (367)^{2} (373)^{2} (379)^{2} (383)^{2} (389)^{2} (397)^{2} (401)^{2} (409)^{2} (419)^{2} (421)^{2} (431)^{2} (433)^{2} (439)^{2} (443)^{2} (449)^{2} (457)^{2} (461)^{2} (463)^{2} (467)^{2} (479)^{2} (487)^{2} (491)^{2} (499)^{2} (503)^{2} (509)^{2} (521) (523)(541)(547)(557)(563)(569)(571)(577)(587)(593)(599)(601)(607)(613) (617)(619)(631)(641)(643)(647)(653)(659)(661)(673)(677)(683)(691)(701) (709)(719)(727)(733)(739)(743)(751)(757)(761)(769)(773)(787)(797)(809) (811)(821)(823)(827)(829)(839)(853)(857)(859)(863)(877)(881)(883)(887) (907)(911)(919)(929)(937)(941)(947)(953)(967)(971)(977)(983)(991)(997) (1009)(1013)(1019)(1021)$

$t := 7.967$

Ordre de grandeur de 1000!

```
> evalf(1000!) ;
```

$4023872601\ 10^{2568}$

Combien de lignes (de 80 caractères) d'affichage ?

```
> iquo(2568, 80) ;
```

32

```
> irem(iquo(1000!, 10^249), 10) ;
```

2

```
> irem(iquo(1000!, 10^248), 10) ;
```

0

Quels sont les trois derniers chiffres de la somme (infinie) des factorielles ?

```
> sum(k!, k=0..13) ;
```

6749977114

```
> sum(k!, k=0..14) ;
```

93928268314

```
> sum(k!, k=0..100) ;
```

94269001683709979260859834124473539872070722613982672442938359305624678223 4\
79506023400294093599136466986609124347432647622826870038220556442336528920 4\
20940314

```
>
```

Combinatoire

Groupe calcul formel - IREM de Grenoble

Fiche n° 3

```
> restart;
```

Les C(n,p) existent...

```
> binomial(5,1) ; binomial(12,5) ;
```

$$5$$

$$792$$

séquence de C(n, p) : la ligne 10 du triangle de Pascal

```
> seq( binomial(10, p), p = 0..10);
```

$$1, 10, 45, 120, 210, 252, 210, 120, 45, 10, 1$$

```
> 10!/(5!*5!);
```

$$252$$

Le triangle de Pascal

```
> limax := 16 :
> for n from 0 to limax do seq( binomial(n, p), p = 0..n) od ;
```

$$1$$
$$1, 1$$
$$1, 2, 1$$
$$1, 3, 3, 1$$
$$1, 4, 6, 4, 1$$
$$1, 5, 10, 10, 5, 1$$
$$1, 6, 15, 20, 15, 6, 1$$
$$1, 7, 21, 35, 35, 21, 7, 1$$
$$1, 8, 28, 56, 70, 56, 28, 8, 1$$
$$1, 9, 36, 84, 126, 126, 84, 36, 9, 1$$
$$1, 10, 45, 120, 210, 252, 210, 120, 45, 10, 1$$
$$1, 11, 55, 165, 330, 462, 462, 330, 165, 55, 11, 1$$
$$1, 12, 66, 220, 495, 792, 924, 792, 495, 220, 66, 12, 1$$
$$1, 13, 78, 286, 715, 1287, 1716, 1716, 1287, 715, 286, 78, 13, 1$$
$$1, 14, 91, 364, 1001, 2002, 3003, 3432, 3003, 2002, 1001, 364, 91, 14, 1$$
$$1, 15, 105, 455, 1365, 3003, 5005, 6435, 6435, 5005, 3003, 1365, 455, 105, 15, 1$$
$$1, 16, 120, 560, 1820, 4368, 8008, 11440, 12870, 11440, 8008, 4368, 1820, 560, 120, 16, 1$$

```
> n := 10 ; sum(binomial(n,j), j=0..n);
```

$$n := 10$$

$$1024$$

```
> n := 20 ; sum(binomial(n,j), j=0..n); 2^20 ;
```

$$n := 20$$

$$1048576$$

$$1048576$$

```
> sum(binomial(l,j), j=0..l);
```

$$2^l$$

```
> t:=time() ; n:= 100 ; sum(binomial(n,j), j=0..n); time()-t ;
```

$$t := 44.060$$

$$n := 100$$

$$1267650600228229401496703205376$$

$$.302$$

Fabriquons nos propres fonctions

```
> Cnp := (n, p) -> n!/(p!*(n-p)!) ;
```

$$Cnp := (n, p) \to \frac{n!}{p!\,(n-p)!}$$

```
> Cnp(10, 5) ;
```

$$252$$

```
> t:=time() ; n:= 100 ; sum(Cnp(n,j), j=0..n); time()-t ;
```

$$t := 44.536$$

$$n := 100$$

$$1267650600228229401496703205376$$

$$.206$$

```
> n := 'n'  : sum(Cnp(n,j), j=0..n);
```

$$2^n$$

```
> Anp := (n, p) -> n! / (n-p) ! ; Anp(6,6) ;
```

$$Anp := (n, p) \to \frac{n!}{(n-p)!}$$

$$720$$

```
> expand((x+1)^10) ;
```

$$x^{10} + 10\,x^9 + 45\,x^8 + 120\,x^7 + 210\,x^6 + 252\,x^5 + 210\,x^4 + 120\,x^3 + 45\,x^2 + 10\,x + 1$$

```
> expand((a+b)^15) ;
```

$$a^{15} + 15\,b\,a^{14} + 105\,b^2\,a^{13} + 455\,b^3\,a^{12} + 1365\,b^4\,a^{11} + 3003\,b^5\,a^{10} + 5005\,b^6\,a^9 + 6435\,b^7\,a^8 + 6435\,b^8\,a^7 + 5005\,b^9\,a^6 + 3003\,b^{10}\,a^5 + 1365\,b^{11}\,a^4 + 455\,b^{12}\,a^3 + 105\,b^{13}\,a^2 + 15\,b^{14}\,a + b^{15}$$

```
> f := (a, b, n) -> (a + b)^n ;
```

$$f := (a, b, n) \to (a+b)^n$$

```
> expand(f(x, -y, 5)) ;
```

$$x^5 - 5\,x^4\,y + 10\,x^3\,y^2 - 10\,x^2\,y^3 + 5\,x\,y^4 - y^5$$

```
>
```

Probabilités

Groupe calcul formel - IREM de Grenoble

Fiche n° 4

```
> restart;
```

Une urne contient 6 boules marquées "a", "b",... "f". On tire deux boules successivement avec remise dans cette urne. Soit Omega l'univers ; dans ce cas Omega=B×B.

```
> B:={a,b,c,d,e,f} :
> Omega :={seq(seq([B[i],B[j]],j=1..nops(B)),i=1..nops(B))};
```

$\Omega := \{[c, f], [c, e], [c, a], [f, f], [f, e], [f, a], [f, b], [f, d], [f, c], [e, f], [e, e], [e, a], [e, b], [e, d], [e, c], [a, f], [a, e], [a, a], [a, b], [a, d], [a, c], [b, f], [b, e], [b, a], [b, b], [b, d], [b, c], [d, f], [d, e], [d, a], [d, b], [d, d], [d, c], [c, b], [c, d], [c, c]\}$

```
> nbev := nops(Omega);
```

$$nbev := 36$$

Evénement 1 : la première boule tirée est la boule "e".

```
> Fav1:=select(Eventualite->Eventualite[1]=e,Omega);
```

$$Fav1 := \{[e, f], [e, e], [e, a], [e, b], [e, d], [e, c]\}$$

```
> P:=proc(Evenement);
> RETURN(nops(Evenement)/nbev);
> end:
> Proba1:=P(Fav1);
```

$$Proba1 := \frac{1}{6}$$

Les 6 boules sont colorées.

```
> a[Couleur]:=rouge:b[Couleur]:=jaune:c[Couleur]:=vert:d[Couleur]:
  =rouge:e[Couleur]:=rouge:f[Couleur]:=jaune:
```

Evénement 2 : les 2 boules tirées sont de même couleur.

```
> Fav2:=select(Eventualite->(Eventualite[1][Couleur]=Eventualite[2
  ][Couleur]),Omega);
```

$Fav2 := \{[f, f], [f, b], [e, e], [e, a], [e, d], [a, e], [a, a], [a, d], [b, f], [b, b], [d, e], [d, a], [d, d], [c, c]\}$

```
> Proba2:=P(Fav2);
```

$$Proba2 := \frac{7}{18}$$

Evénement 3 : la première boule tirée est "e" ET les 2 boules sont de même couleur.

```
> Fav3:=Fav1 intersect Fav2;
```

$$Fav3 := \{[e, e], [e, a], [e, d]\}$$

```
> Proba3:=P(Fav3);
```

$$Proba3 := \frac{1}{12}$$

Les événements 1 et 2 sont-ils indépendants ?

```
> Independance := evalb(Proba3=Proba1*Proba2);
```

$Independance := false$

Evénement 4 : la deuxième boule est rouge.

```
> Fav4:=select(Eventualite->Eventualite[2][Couleur]=rouge,Omega);
```

$Fav4 := \{[c, e], [c, a], [f, e], [f, a], [f, d], [e, e], [e, a], [e, d], [a, e], [a, a], [a, d], [b, e], [b, a], [b, d], [d, e], [d, a], [d, d], [c, d]\}$

```
> Proba4:=P(Fav4);
```

$$Proba4 := \frac{1}{2}$$

Les événements 1 et 4 sont-ils indépendants ?

```
> Fav5:=Fav1 intersect Fav4;
```

$$Fav5 := \{[e, e], [e, a], [e, d]\}$$

```
> Proba5:=P(Fav5);
```

$$Proba5 := \frac{1}{12}$$

```
> Independance := evalb(Proba5=Proba1*Proba4);
```

$$Independance := true$$

SIMULATION : on effectue 1000 tirages et on calcule les fréquences observées.

```
> tirage :=rand(1..6) :Digits:=3:
  Exp := NULL: nbexp:= 100:
  for i from 1 to nbexp do
  Exp :=Exp ,[B[tirage()],B[tirage()]]:
  od:
  Exp:=[Exp]:
> Exp; # pour voir
```

$[[b, a], [b, c], [d, a], [c, a], [e, e], [e, b], [b, a], [a, e], [f, b], [b, c], [f, f], [f, e], [b, e], [f, a], [c, a], [c, b], [a, c], [f, c], [a, f], [b, c], [b, a], [e, d], [f, d], [b, b], [e, c], [c, f], [e, e], [b, e], [b, b], [e, c], [f, f], [b, b], [e, b], [f, b], [d, a], [b, a], [b, b], [a, c], [c, d], [c, f], [a, e], [e, d], [f, d], [e, d], [c, d], [b, b], [c, b], [c, e], [a, e], [f, e], [b, d], [b, c], [f, e], [c, a], [d, f], [d, b], [e, a], [f, d], [f, c], [c, d], [c, d], [b, e], [a, d], [a, d], [d, e], [a, b], [f, b], [f, a], [b, a], [d, d], [d, c], [f, b], [e, b], [b, a], [a, d], [b, d], [f, a], [b, b], [c, d], [f, f], [b, a], [c, c], [c, f], [b, b], [e, e], [e, e], [c, b], [b, b], [c, c], [a, a], [a, c], [f, d], [e, a], [d, b], [a, e], [b, c], [f, f], [a, b], [b, e], [a, f]]$

```
> fav1:=select(Eventualite->Eventualite[1]=e,Exp);
```

$fav1 := [[e, e], [e, b], [e, d], [e, c], [e, e], [e, c], [e, b], [e, d], [e, d], [e, a], [e, b], [e, e], [e, e], [e, a]]$

```
> Frequ1:=evalf(nops(fav1)/nbexp); evalf(Proba1);
```

$$Frequ1 := .140$$

$$.167$$

```
> fav2:=select(Eventualite->Eventualite[1][Couleur]=Eventualite[2]
  [Couleur],Exp);
```

$fav2 := [[d, a], [e, e], [a, e], [f, b], [f, f], [e, d], [b, b], [e, e], [b, b], [f, f], [b, b], [f, b], [d, a], [b, b], [a, e], [e, d], [e, d], [b, b], [a, e], [e, a], [a, d], [a, d], [d, e], [f, b], [d, d],$

$[f, b], [a, d], [b, b], [f, f], [c, c], [b, b], [e, e], [e, e], [b, b], [c, c], [a, a], [e, a], [a, e],$
$[f, f]]$

```
> Frequ2:=evalf(nops(fav2)/nbexp); evalf(Proba2);
```

$$Frequ2 := .390$$

$$.389$$

```
> fav3:=select(Eventualite->Eventualite[1]=e and
  Eventualite[1][Couleur]=Eventualite[2][Couleur],Exp);
```

$$fav3 := [[e, e], [e, d], [e, e], [e, d], [e, d], [e, a], [e, e], [e, e], [e, a]]$$

```
> Frequ3:=evalf(nops(fav3)/nbexp); evalf(Proba3);
```

$$Frequ3 := .0900$$

$$.0833$$

```
> fav4:=select(Eventualite->Eventualite[2][Couleur]=rouge,Exp);
```

$fav4 := [[b, a], [d, a], [c, a], [e, e], [b, a], [a, e], [f, e], [b, e], [f, a], [c, a], [b, a], [e, d],$
$[f, d], [e, e], [b, e], [d, a], [b, a], [c, d], [a, e], [e, d], [f, d], [e, d], [c, d], [c, e], [a, e],$
$[f, e], [b, d], [f, e], [c, a], [e, a], [f, d], [c, d], [c, d], [b, e], [a, d], [a, d], [d, e], [f, a],$
$[b, a], [d, d], [b, a], [a, d], [b, d], [f, a], [c, d], [b, a], [e, e], [e, e], [a, a], [f, d], [e, a],$
$[a, e], [b, e]]$

```
> Frequ4:=evalf(nops(fav4)/nbexp); evalf(Proba4);
```

$$Frequ4 := .530$$

$$.500$$

```
> fav5:=select(Eventualite->Eventualite[1]=e and
  Eventualite[2][Couleur]=rouge,Exp);
```

$$fav5 := [[e, e], [e, d], [e, e], [e, d], [e, d], [e, a], [e, e], [e, e], [e, a]]$$

```
> Frequ5:=evalf(nops(fav5)/nbexp); evalf(Proba5);
```

$$Frequ5 := .0900$$

$$.0833$$

```
>
```

Arithmétique

Groupe calcul formel - IREM de Grenoble

Fiche n° 5

```
> restart;
```

Division euclidienne, pgcd, ppcm... (initiales en anglais)

```
> iquo(54, 7) ; quotient := iquo ;
```

$$7$$

$$quotient := iquo$$

```
> irem(54, 7) ; reste := irem ;
```

$$5$$

$$reste := irem$$

```
> igcd(124, 788) ; 124/788 ; pgcd := igcd ;
```

$$4$$

$$\frac{31}{197}$$

$$pgcd := igcd$$

```
> igcdex(124, 788, 's', 't') ; s ; t ; s*124+t*788 ; bezout :=
  igcdex ;
```

$$4$$

$$89$$

$$-14$$

$$4$$

$$bezout := igcdex$$

```
> ilcm(10,35) ; 1/10+1/35 ; ppcm := ilcm ;
```

$$70$$

$$\frac{9}{70}$$

$$ppcm := ilcm$$

Décomposition en facteurs premiers

```
> ifactor(70) ;
```

$$(2)(5)(7)$$

```
> ifactor(100!) ;
```

$$(2)^{97}(3)^{48}(5)^{24}(7)^{16}(11)^{9}(13)^{7}(17)^{5}(19)^{5}(23)^{4}(29)^{3}(31)^{3}(37)^{2}(41)^{2}(43)^{2}(47)^{2}$$
$$(53)(59)(61)(67)(71)(73)(79)(83)(89)(97)$$

```
> ifactor(2^20-1) ;
```

$$(3)(5)^{2}(11)(31)(41)$$

```
> isprime(2^(2^7)-1) ;
```

$$false$$

```
> ifactor(2^(2^7)-1) ;
```

(3) (5) (17) (257) (641) (67280421310721) (6700417) (65537) (274177)

```
> isprime(41) ;
```

true

Séquence de nombres premiers

```
> n := 1000 :
> c := 0 : for i from 3 by 2 to n do if isprime(i) then c := c+1 :
  t[c] := i fi od;
> seq(t[j], j = 1..c) ;
```

3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113, 127, 131, 137, 139, 149, 151, 157, 163, 167, 173, 179, 181, 191, 193, 197, 199, 211, 223, 227, 229, 233, 239, 241, 251, 257, 263, 269, 271, 277, 281, 283, 293, 307, 311, 313, 317, 331, 337, 347, 349, 353, 359, 367, 373, 379, 383, 389, 397, 401, 409, 419, 421, 431, 433, 439, 443, 449, 457, 461, 463, 467, 479, 487, 491, 499, 503, 509, 521, 523, 541, 547, 557, 563, 569, 571, 577, 587, 593, 599, 601, 607, 613, 617, 619, 631, 641, 643, 647, 653, 659, 661, 673, 677, 683, 691, 701, 709, 719, 727, 733, 739, 743, 751, 757, 761, 769, 773, 787, 797, 809, 811, 821, 823, 827, 829, 839, 853, 857, 859, 863, 877, 881, 883, 887, 907, 911, 919, 929, 937, 941, 947, 953, 967, 971, 977, 983, 991, 997

Changements de base

```
> convert(456, binary) ; convert(456, base, 2) ;
```

111001000

[0, 0, 0, 1, 0, 0, 1, 1, 1]

```
> convert(456, octal) ; convert(456, base, 8) ;
```

710

[0, 1, 7]

```
> convert([1, 1, 1, 0, 1], base, 2, 10) ;
```

[3, 2]

```
> convert(convert(23, binary), decimal, binary) ;
```

23

```
>  a:= convert(1789, base, 16) ; convert(a, base, 16, 10) ;
```

$a := [13, 15, 6]$

[9, 8, 7, 1]

Période des restes

```
> seq(10^p mod 17 , p=0..15) ;
```

1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12

```
> rp := proc(n, q) local p, l ;
> l := seq(10^p mod n, p=(n-1)*q..(n-1)*(q+1)-1) ;
> end ;
```

$rp := \mathbf{proc}(n, q)\ \mathbf{local}\ p, l;\ l := \mathrm{seq}(\,10\text{^}p \bmod n, p = (n-1)*q\ ..\ (n-1)*(q+1)-1\,)\ \mathbf{end}$

```
> rp(17, 0) ;
```

1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12

```
>  for q from 0 to 4 do rp(17, q); od ;
```

1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12

$$1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12$$
$$1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12$$
$$1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12$$
$$1, 10, 15, 14, 4, 6, 9, 5, 16, 7, 2, 3, 13, 11, 8, 12$$

```
> a := 89 ; p := 17 ; reste(a^(p-1), p) ;
```

$$a := 89$$
$$p := 17$$
$$1$$

```
> divisible := proc(a, b) local r :
> r := a mod b :
> RETURN(evalb(r=0)) :
> end ;
```

$$\mathit{divisible} := \textbf{proc}(a, b)\ \textbf{local}\ r;\ r := a \bmod b;\ \mathrm{RETURN}(\mathrm{evalb}(r = 0))\ \textbf{end}$$

```
> divisible(51,17) ;
```

true

```
> divisible(a^p-a, p) ;
```

true

```
>
```

Polynômes

Groupe calcul formel - IREM de Grenoble

Fiche n° 6

```
> restart;
> expand((x-3)^4) ;
```

$$x^4 - 12\,x^3 + 54\,x^2 - 108\,x + 81$$

```
> factor(") ;
```

$$(x-3)^4$$

```
> expand((x+2)^20) ;
```

$$x^{20} + 40\,x^{19} + 760\,x^{18} + 9120\,x^{17} + 77520\,x^{16} + 496128\,x^{15} + 2480640\,x^{14} + 9922560\,x^{13} + 32248320\,x^{12} + 85995520\,x^{11} + 189190144\,x^{10} + 343982080\,x^9 + 515973120\,x^8 + 635043840\,x^7 + 635043840\,x^6 + 508035072\,x^5 + 317521920\,x^4 + 149422080\,x^3 + 49807360\,x^2 + 10485760\,x + 1048576$$

```
> factor(x^3+125) ; factor(x^4+2500) ; factor(x^5+1) ;
```

$$(x+5)\,(x^2 - 5\,x + 25)$$

$$(x^2 - 10\,x + 50)\,(x^2 + 10\,x + 50)$$

$$(x+1)\,(x^4 - x^3 + x^2 - x + 1)$$

```
> p := x -> product(x+i!, i=1..20) ;
```

$$p := x \to \prod_{i=1}^{20} (x + i!)$$

```
> expand(p(x)) ;
```

2125189815046852759395278881794666693166559689452392888997366823856489273601\
8553984656205904300470389140992834438302374297600000000 x^{10} + 527154145479417871\
7620332876977236988373225218954544722776530101469453956135864846872915562081\
520891164841740387483648000000 x^{11} + 10915459152802424702133470060021457866574\
68015416707747222949677044196129442264619388955273015159070602035200000 x^{12} +
174088378793822747493593274565205209984265428816493179621700510592038791985\
845844497874437897831219200 x^{13} + 1985228837776275514288895048037772454332004\
668692848878753779305672665536147702122043924480 x^{14} + 15104633683060648732151\
0221579205955521736886414117608011729578102887937351555840 x^{15}
+ 718740745560998393795808305626478506882464107851974816669172659648 x^{16}
+ 2012862082698591716096362592063659200321327531519372 x^{17}
+ 31327385145381966234448325226478307 6 x^{18} + 2561327494111820313 x^{19} + x^{20} + 4603\
7776061111865953673723942334983164129050825944843926720198933140216383147331\
88264234198210564706933487746170961841984397758211031040000000000000000000 x^5

$+$ 6136595629161610435039567610328301296017147463783815749439997349830575692\
0488613802574011913908942351056961326961974947502626913124352000000000000000\
0 x^6 + 1183949087591171481855055193602517800876965553393529151865273383415102\
19479964652932741403067225800962643871878060947961220970840064000000000000000\
0 x^7 + 2877229875998384837410849760702847844701925249492010350590521504572621\
752377303930375458721259901665377402288117642280882665422848000000000000 x^8 +
7806844422012808258548267155811117591192547396374774021798566290595222072 48\
897535040298949179771908227212859703297834890559488000000000 x^9 + 21898132645\
8442174969113326542894376796546861997227260982282573122357584338725036688 22\
568323179215330593924529580562707256442880000000000000000000000000000 x + 1\
0659934197659712544430246181126421855961312796235049183524438926206301468 44\
6888391162559600621762561770028158943581688935795916800000000000000000000000\
000000 x^2 + 58922187226514716773647344330812514576716449858102225622424063328\
178154428748527741214550139551290904657624882163092515125778836357120000000\
0000000000000 x^4 + 15644670986386771721293463417001005003847660985933287148 7\
8480372541242993269594209404858144582479234694395043078714912327073708113 92\
0000000000000000000000 x^3 + 127442031238161076418723266959124572809405357 64\
803996814646074786210860158451928733118473415711895225119341900569911702323\
200000000000000000000000000000000

```
> factor(p(x));
```

$$(x+1)(x+2)(x+6)(x+24)(x+120)(x+720)(x+5040)(x+40320)(x+362880)$$
$$(x+3628800)(x+39916800)(x+479001600)(x+6227020800)(x+87178291200)$$
$$(x+1307674368000)(x+20922789888000)(x+355687428096000)$$
$$(x+6402373705728000)(x+121645100408832000)(x+2432902008176640000)$$

```
> p2:=(x, y) -> 9*(2+x)*(x+y) + (x+y) ^2 ;
```

$$p2 := (x,y) \rightarrow 9\,(2+x)\,(x+y) + (x+y)^2$$

```
> expand(p2(x,y)^3) ;
```

$$5400\,x^5 + 1000\,x^6 + 9720\,x^4 + 5832\,x^3 + 5832\,y^3 + 972\,y^4 + 54\,y^5 + y^6 + 30132\,x^3\,y$$
$$+ 17496\,x^2\,y + 32076\,x^2\,y^2 + 17496\,x\,y^2 + 12636\,x\,y^3 + 8802\,x^2\,y^3 + 393\,x^2\,y^4 + 3300\,x^5\,y$$
$$+ 17280\,x^4\,y + 3930\,x^4\,y^2 + 19494\,x^3\,y^2 + 1991\,x^3\,y^3 + 1242\,x\,y^4 + 33\,x\,y^5$$

```
> collect(",x) ;
```

$$1000\,x^6 + (3300\,y + 5400)\,x^5 + (9720 + 3930\,y^2 + 17280\,y)\,x^4$$
$$+ (19494\,y^2 + 5832 + 30132\,y + 1991\,y^3)\,x^3 + (393\,y^4 + 32076\,y^2 + 17496\,y + 8802\,y^3)\,x^2$$
$$+ (33\,y^5 + 17496\,y^2 + 12636\,y^3 + 1242\,y^4)\,x + 5832\,y^3 + 972\,y^4 + 54\,y^5 + y^6$$

```
> factor(") ;
```

$$(x+y)^3\,(10\,x+y+18)^3$$

```
> r:=x->sum(binomial(10, p)*x^p*3^(10-p), p=0..10) ;
```

$$r := x \to \sum_{p=0}^{10} \mathrm{binomial}(10, p)\, x^p\, 3^{(10-p)}$$

```
> expand(r(x)) ;
```

$$59049 + 196830\, x + 295245\, x^2 + 262440\, x^3 + 153090\, x^4 + 61236\, x^5 + 17010\, x^6 + 3240\, x^7 + 405\, x^8 + 30\, x^9 + x^{10}$$

```
> factor(") ;
```

$$(x+3)^{10}$$

pgcd, ppcm, bezout...

```
> p1 := x -> (x-3)^3 ;
```

$$p1 := x \to (x-3)^3$$

```
> p2 := x -> (x-3)*(x+2)^2 ;
```

$$p2 := x \to (x-3)\,(2+x)^2$$

```
> d := gcd(p1(x), p2(x)) ;
```

$$d := x - 3$$

```
> gcd(p1(x), p2(x), 'u', 'v') ; u ; v ;
```

$$x - 3$$

$$(x-3)^2$$

$$(x+2)^2$$

```
>  factor(u*d) ; factor( v*d) ;
```

$$(x-3)^3$$

$$(x-3)\,(x+2)^2$$

```
> gcdex(p1(x), p2(x), x, 'u', 'v'); u, v ;
```

$$x - 3$$

$$\frac{9}{125} + \frac{2}{125}\,x,\ \frac{11}{125} - \frac{2}{125}\,x$$

```
> simplify(u*p1(x)+v*p2(x)) ;
```

$$x - 3$$

```
> lcm(p1(x), p2(x)) ;
```

$$x^5 - 5\,x^4 - 5\,x^3 + 45\,x^2 - 108$$

```
>
```

Equations-Inéquations

Groupe calcul formel - IREM de Grenoble

Fiche n° 7

```
> restart;
```

Remarque :

1° l'utilisation des ensembles, obligatoire pour les systèmes, peut être utile même pour une seule équation ou une seule inéquation,

2° commencer la feuille par restart, permet de s'assurer que les inconnues sont des variables muettes,

3° >_EnvExplicit := true ; force Maple à exprimer les solutions avec des radicaux, jusqu'au degré 4, pour toute une session,

4° >interface(labelling=false) ; force Maple à remplacer les %1 %2... par leurs valeurs pour toute une session.

```
> Sol:=solve(3*x+Pi=0,x) ;
```

$$Sol := -\frac{1}{3}\pi$$

```
> Sol:=solve((2+3*I)*x-4*(2-I)=0,x) ;
```

$$Sol := \frac{4}{13} - \frac{32}{13} I$$

```
> Sol:=solve(x^2+3*x-4=0,x) ;
```

$$Sol := -4, 1$$

```
> Sol:=solve(4*x^2-7*x-45=0,x);
```

$$Sol := \frac{7}{8} + \frac{1}{8}\sqrt{769}, \frac{7}{8} - \frac{1}{8}\sqrt{769}$$

```
> Sol:=solve(-x^2+12*x-36=0,x) ;
```

$$Sol := 6, 6$$

```
> Sol:=solve(x^2+x+1=0,x) ;
```

$$Sol := -\frac{1}{2} + \frac{1}{2} I\sqrt{3}, -\frac{1}{2} - \frac{1}{2} I\sqrt{3}$$

```
> Sol:=solve(x^3=1,x) ;
```

$$Sol := 1, -\frac{1}{2} + \frac{1}{2} I\sqrt{3}, -\frac{1}{2} - \frac{1}{2} I\sqrt{3}$$

```
> Sol:=solve(x^6=1,x) ;
```

$$Sol := 1, -1, -\frac{1}{2} + \frac{1}{2} I\sqrt{3}, -\frac{1}{2} - \frac{1}{2} I\sqrt{3}, \frac{1}{2} - \frac{1}{2} I\sqrt{3}, \frac{1}{2} + \frac{1}{2} I\sqrt{3}$$

```
> Sol:=solve(x^3-3*x=0,x) ;
```

$$Sol := 0, \sqrt{3}, -\sqrt{3}$$

```
> Sol:=solve(x^3+x^2+1=0,x) ;
```

$$Sol := -\frac{1}{6}(116+12\sqrt{93})^{1/3} - \frac{2}{3}\frac{1}{(116+12\sqrt{93})^{1/3}} - \frac{1}{3}, \frac{1}{12}(116+12\sqrt{93})^{1/3} + \frac{1}{3}\frac{1}{(116+12\sqrt{93})^{1/3}} - \frac{1}{3} + \frac{1}{2}I\sqrt{3}\left(-\frac{1}{6}(116+12\sqrt{93})^{1/3} + \frac{2}{3}\frac{1}{(116+12\sqrt{93})^{1/3}}\right),$$
$$\frac{1}{12}(116+12\sqrt{93})^{1/3} + \frac{1}{3}\frac{1}{(116+12\sqrt{93})^{1/3}} - \frac{1}{3} - \frac{1}{2}I\sqrt{3}\left(-\frac{1}{6}(116+12\sqrt{93})^{1/3} + \frac{2}{3}\frac{1}{(116+12\sqrt{93})^{1/3}}\right)$$

```
> interface(labelling=false);
> Sol:=solve(x^3+x^2+1=0,x) ;
```

$$Sol := -\frac{1}{6}(116+12\sqrt{93})^{1/3} - \frac{2}{3}\frac{1}{(116+12\sqrt{93})^{1/3}} - \frac{1}{3}, \frac{1}{12}(116+12\sqrt{93})^{1/3} + \frac{1}{3}\frac{1}{(116+12\sqrt{93})^{1/3}} - \frac{1}{3} + \frac{1}{2}I\sqrt{3}\left(-\frac{1}{6}(116+12\sqrt{93})^{1/3} + \frac{2}{3}\frac{1}{(116+12\sqrt{93})^{1/3}}\right),$$
$$\frac{1}{12}(116+12\sqrt{93})^{1/3} + \frac{1}{3}\frac{1}{(116+12\sqrt{93})^{1/3}} - \frac{1}{3} - \frac{1}{2}I\sqrt{3}\left(-\frac{1}{6}(116+12\sqrt{93})^{1/3} + \frac{2}{3}\frac{1}{(116+12\sqrt{93})^{1/3}}\right)$$

```
> evalf(") ;
```

$$-1.465571232, .2327856159 - .7925519930\,I, .2327856159 + .7925519930\,I$$

```
> eq := x^4-5*x^2+6*x=2;
```

$$eq := x^4 - 5\,x^2 + 6\,x = 2$$

```
> Sol:=solve({eq},x);
```

$$Sol := \{x = -1 + \sqrt{3}\}, \{x = -1 - \sqrt{3}\}, \{x = 1\}, \{x = 1\}$$

```
> evalf(");
```

$$\{x = .732050808\}, \{x = -2.732050808\}, \{x = 1.\}, \{x = 1.\}$$

```
> for i from 1 to 4 do x[i]:=subs( Sol[i], x ) od;
```

$$x_1 := -1 + \sqrt{3}$$
$$x_2 := -1 - \sqrt{3}$$
$$x_3 := 1$$
$$x_4 := 1$$

Inéquations

```
> Sol:=solve(2*x+3<0,x);
```

$$Sol := \mathrm{RealRange}\left(-\infty, \mathrm{Open}\left(\frac{-3}{2}\right)\right)$$

```
> Sol:=solve({2*x+3<0},x); # c'est mieux, non ?
```

$$Sol := \{x < \frac{-3}{2}\}$$

```
> Sol:=solve( {x^2+5*x+4<0},x) ;
```

$$Sol := \{x < -1, -4 < x\}$$

```
> Sol:=solve({x^2+5*x+4>=0},x) ;
```

$$Sol := \{x \leq -4\}, \{-1 \leq x\}$$

```
> Sol:=solve({x^2+8*x+16> 0},x) ;
```

$$Sol := \{x < -4\}, \{-4 < x\}$$

```
> Sol:=solve({(x^2+5*x+4)/(3+x)<0},x) ;
```

$$Sol := \{x < -4\}, \{x < -1, -3 < x\}$$

Systèmes

```
> Sol:=solve({(x+3)/2-(y-1)/4=2,(x-2)/2-(y+3)/4=0},{x,y}); # Maple
  ne trouve pas de solution
```

$$Sol :=$$

```
> Sol:=solve({x*sqrt(3)+y=1,3*x+y*sqrt(3)=sqrt(3)},{x,y}); # x est
  choisi comme paramètre
```

$$Sol := \{y = -x\sqrt{3} + 1, x = x\}$$

```
> Sol:=solve({3/(2*x)+2/(5*y)=1/2,5/(2*x)-7/(3*y)=4/3},{x,y});
```

$$Sol := \{y = -6, x = \frac{45}{17}\}$$

```
> subs(Sol,[x,y]); # retourne le couple solution, sans pour autant
  assigner x et y
```

$$\left[\frac{45}{17}, -6\right]$$

```
> x; y; # la preuve !
```

$$x$$

$$y$$

```
> Sol:=solve({2*x+2=0},x);
```

$$Sol := \{x = -1\}$$

```
> subs(Sol,x); # retourne la solution, sans pour autant assigner x
```

$$-1$$

```
> Sol:=solve({x^2+4*y^2=4,3*x^2-8*y^2=-3},{x,y});
```

$$Sol := \{x = 1, y = \frac{1}{2}\mathrm{RootOf}(\_Z^2 - 3)\}, \{y = \frac{1}{2}\mathrm{RootOf}(\_Z^2 - 3), x = -1\}$$

```
> _EnvExplicit:=true:
> Sol:=solve({x^2+4*y^2=4,3*x^2-8*y^2=-3},{x,y});
```

$$Sol := \{y = \frac{1}{2}\sqrt{3}, x = 1\}, \{y = -\frac{1}{2}\sqrt{3}, x = 1\}, \{y = \frac{1}{2}\sqrt{3}, x = -1\}, \{y = -\frac{1}{2}\sqrt{3}, x = -1\}$$

```
> Sol:=solve({x+2*y=5,2*x^2-3*y^2=-11},{x,y});
```

$$Sol := \{x = -3 - \frac{2}{5}\sqrt{95}, y = 4 + \frac{1}{5}\sqrt{95}\}, \{x = -3 + \frac{2}{5}\sqrt{95}, y = 4 - \frac{1}{5}\sqrt{95}\}$$

```
> Sol:=solve({x+y+z=2,y+z+t=-1,z+t+x=1,t+x+y=-2},{x,y,z,t});
```

$$Sol := \{z=2, y=-1, x=1, t=-2\}$$

```
> SystEq:={2*x-y=m,2*m*x+2*y=1};
  # m désigne un paramètre réel ; la discussion doit être faite
  par le mathématicien !
```

$$SystEq := \{2\,x-y=m,\ 2\,m\,x+2\,y=1\}$$

```
> Sol:=solve(SystEq,{x,y}); # pour m différent de -2
```

$$Sol := \{x=\frac{1}{2}\,\frac{2\,m+1}{m+2}, y=-\frac{-1+m^2}{m+2}\}$$

```
> Sol:=solve(subs(m=-2,SystEq),{x,y}); # pour m égal à -2
```

$$Sol :=$$

```
> m; # m désigne toujours le paramètre réel
```

$$m$$

```
> eqs:={cos(a)+cos(x+a)+cos(y+a)=0,sin(a)+sin(x+a)+sin(y+a)=0};
```

$$eqs := \{\cos(a)+\cos(x+a)+\cos(y+a)=0, \sin(a)+\sin(x+a)+\sin(y+a)=0\}$$

```
> Sol:=solve(eqs,{x,y});
```

$$Sol := \{x=-a+\arctan\left(-\frac{1}{2}\,\frac{1+\cos(a)\,(-\cos(a)+\sqrt{\cos(a)^2+4\sin(a)^2-1}\,)}{\sin(a)},\right.$$
$$\left.-\frac{1}{2}\cos(a)+\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\right), y=\arctan\left(\right.$$
$$\frac{1}{2}\,\frac{-2\sin(a)^2+1+\cos(a)\,(-\cos(a)+\sqrt{\cos(a)^2+4\sin(a)^2-1}\,)}{\sin(a)},$$
$$\left.-\frac{1}{2}\cos(a)-\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\right)-a\}, \{y=\arctan\left(\right.$$
$$\frac{1}{2}\,\frac{-2\sin(a)^2+1+\cos(a)\,(-\cos(a)-\sqrt{\cos(a)^2+4\sin(a)^2-1}\,)}{\sin(a)},$$
$$\left.-\frac{1}{2}\cos(a)+\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\right)-a, x=-a+\arctan\left(\right.$$
$$-\frac{1}{2}\,\frac{1+\cos(a)\,(-\cos(a)-\sqrt{\cos(a)^2+4\sin(a)^2-1}\,)}{\sin(a)},$$
$$\left.-\frac{1}{2}\cos(a)-\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\right)\}$$

```
> Sol:=solve(eqs,{x,y});
```

$$Sol := \{x=-a+\arctan\left(-\frac{1}{2}\,\frac{1+\cos(a)\,(-\cos(a)+\sqrt{\cos(a)^2+4\sin(a)^2-1}\,)}{\sin(a)},\right.$$
$$\left.-\frac{1}{2}\cos(a)+\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\right), y=\arctan\left(\right.$$
$$\frac{1}{2}\,\frac{-2\sin(a)^2+1+\cos(a)\,(-\cos(a)+\sqrt{\cos(a)^2+4\sin(a)^2-1}\,)}{\sin(a)},$$

$$-\frac{1}{2}\cos(a)-\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\Bigg)-a\}, \{y=\arctan\Bigg($$

$$\frac{1}{2}\frac{-2\sin(a)^2+1+\cos(a)\left(-\cos(a)-\sqrt{\cos(a)^2+4\sin(a)^2-1}\right)}{\sin(a)},$$

$$-\frac{1}{2}\cos(a)+\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\Bigg)-a, x=-a+\arctan\Bigg($$

$$-\frac{1}{2}\frac{1+\cos(a)\left(-\cos(a)-\sqrt{\cos(a)^2+4\sin(a)^2-1}\right)}{\sin(a)},$$

$$-\frac{1}{2}\cos(a)-\frac{1}{2}\sqrt{\cos(a)^2+4\sin(a)^2-1}\Bigg)\}$$

[>

Suites numériques

Groupe calcul formel - IREM de Grenoble

Fiche n° 8

Une suite arithmétique quelconque

```
> restart;
> u:=n->u0+a*n ;
```

$$u := n \to u0 + a\, n$$

Les premiers termes...

```
> u0:=1 : a:=2 : seq(u(n), n=0..20);
```

$$1, 3, 5, 7, 9, 11, 13, 15, 17, 19, 21, 23, 25, 27, 29, 31, 33, 35, 37, 39, 41$$

Limite, somme de 100 termes consécutifs

```
> n:='n' ; limit(u(n), n=infinity) ;
```

$$n := n$$

$$\infty$$

```
> sum(u(n),n=0..99) ; 100^2;
```

$$10000$$

$$10000$$

somme des p+1 premiers termes et limite :

```
> sum(u(n),n=0..p) ;
```

$$\frac{1}{2}p + \frac{1}{2} + \frac{1}{2}(p+1)^2$$

```
> limit(sum(u(n),n=0..p), p=infinity) ;
```

$$\infty$$

Une suite géométrique quelconque

```
> restart;
> u:=n->u0*a^n ;
```

$$u := n \to u0\, a^n$$

Les premiers termes...

```
> u0:=1 : a:=1/2 : seq(u(n), n=0..20);
```

$$1, \frac{1}{2}, \frac{1}{4}, \frac{1}{8}, \frac{1}{16}, \frac{1}{32}, \frac{1}{64}, \frac{1}{128}, \frac{1}{256}, \frac{1}{512}, \frac{1}{1024}, \frac{1}{2048}, \frac{1}{4096}, \frac{1}{8192}, \frac{1}{16384}, \frac{1}{32768}, \frac{1}{65536}, \frac{1}{131072},$$

$\frac{1}{262144}, \frac{1}{524288}, \frac{1}{1048576}$

Limite, somme de 100 termes consécutifs

```
> n:='n' ; limit(u(n), n=infinity) ;
```

$$n := n$$

$$0$$

```
> sum(u(n),n=0..99) ;
```

$$\frac{1267650600228229401496703205375}{633825300114114700748351602688}$$

somme des p+1 premiers termes et limite :

```
> sum(u(n),n=0..p) ;
```

$$-2\left(\frac{1}{2}\right)^{(p+1)}+2$$

```
> limit(sum(u(n),n=0..p), p=infinity) ;
```

$$2$$

```
>
```

Limites

Groupe calcul formel - IREM de Grenoble

Fiche n° 9

```
> restart;
> f:=x->x*(sqrt(x+sqrt(x+1))-sqrt(x+sqrt(x-1)));
```

$$f := x \to x\left(\sqrt{x+\sqrt{x+1}} - \sqrt{x+\sqrt{x-1}}\right)$$

```
> limit(f(x), x=infinity) ;
```

$$\frac{1}{2}$$

```
> g:=x->(x^3+8)/(x+2);
```

$$g := x \to \frac{x^3+8}{x+2}$$

```
> simplify(g(x));
```

$$x^2 - 2x + 4$$

```
> limit(g(x),x=-2) ;
```

$$12$$

```
> h:=x->sqrt(4*x^2-3*x)+2*x;
```

$$h := x \to \sqrt{4x^2 - 3x} + 2x$$

```
> limit(h(x),x=-infinity);
```

$$\frac{3}{4}$$

```
> j:=x->floor(x)+(x-floor(x))^2 ;
```

$$j := x \to \mathrm{floor}(x) + (x - \mathrm{floor}(x))^2$$

floor est la partie entière

```
> plot(j,-5..5);
```




```
> k:=x->x*floor(1/x) ;
```

$$k := x \to x\,\mathrm{floor}\left(\frac{1}{x}\right)$$

```
> plot(k(x),x=0.01..1.1,y=0..1.1, discont=true);
```




```
> limit(k(x), x=infinity) ;
```

$$0$$

```
> limit(k(x), x=0) ;
```

$$\lim_{x \to 0} x \operatorname{floor}\left(\frac{1}{x}\right)$$

```
> limit(sin(x)/x, x=0) ;
```

$$1$$

```
> g := x -> (sin(3*x)-sin(x))/x ;
```

$$g := x \to \frac{\sin(3\,x) - \sin(x)}{x}$$

```
> limit(g(x), x=0) ;
```

$$2$$

```
>
```

Dérivation

Groupe calcul formel - IREM de Grenoble

Fiche n° 10

```
> restart;
> f:= x->x^3-3*x^2+5*x-4 ;
```

$$f := x \to x^3 - 3\,x^2 + 5\,x - 4$$

```
> f1:=D(f) ;
```

$$f1 := x \to 3\,x^2 - 6\,x + 5$$

```
> D(D(f)) ; (D@@2)(f) ;
```

$$x \to 6\,x - 6$$

$$x \to 6\,x - 6$$

```
> (D@@3)(ln) ;
```

$$a \to \frac{2}{a^3}$$

```
> taf:= h->(f(a+h)-f(a))/h;
```

$$taf := h \to \frac{\mathrm{f}(a+h) - \mathrm{f}(a)}{h}$$

```
> limit(taf(h), h=0);
```

$$5 + 3\,a^2 - 6\,a$$

```
> solve(f1(x)=0,x);
```

$$1 + \frac{1}{3}I\sqrt{6},\ 1 - \frac{1}{3}I\sqrt{6}$$

```
> solve(f1(x)>0,x);
```

$$x$$

```
> plot({f,f1},-5..5, -10..10) ;
```




```
> D(h+g) ;
```

$$D(h)+D(g)$$

```
> D(h@g) ;
```

$$(D(h))@g\, D(g)$$

```
> D(h/g);
```

$$\frac{D(h)}{g}-\frac{h\,D(g)}{g^2}$$

```
> va := x -> abs(x) ;
```

$$va := \mathrm{abs}$$

```
> D(va) ;
```

$$a \rightarrow \mathrm{abs}(1, a)$$

```
> abs(1,7) ;
```

$$1$$

```
> abs(1, -7) ;
```

$$-1$$

```
> abs(1,0) ;
Error, (in simpl/abs) abs is not differentiable at 0
> D(exp+tan) ;
```

$$\exp + 1 + \tan^2$$

```
> f := (x, y) -> exp(x*y) ;
```

$$f := (x, y) \rightarrow \mathbf{e}^{(x\,y)}$$

```
> D[1](f) ; D[2](f) ;
```

$$(x, y) \rightarrow y\,\mathbf{e}^{(x\,y)}$$

$$(x, y) \rightarrow x\,\mathbf{e}^{(x\,y)}$$

```
> D[1,2](f) ; D[2, 1](f) ;
```

$$(x, y) \rightarrow \mathbf{e}^{(x\,y)} + x\,y\,\mathbf{e}^{(x\,y)}$$

$$(x, y) \rightarrow \mathbf{e}^{(x\,y)} + x\,y\,\mathbf{e}^{(x\,y)}$$

```
>
```

Fonctions

Groupe calcul formel - IREM de Grenoble

Fiche n° 11

```
> restart;
```

Une fonction, ses fonctions dérivées successives

```
> f:=x->exp(-2*x) ;
```

$$f := x \to e^{(-2\,x)}$$

```
> f1:=D(f) ;
```

$$f1 := x \to -2\,e^{(-2\,x)}$$

```
> f2:=D(D(f)) ;
```

$$f2 := x \to 4\,e^{(-2\,x)}$$

```
> (D@@2)(f) ;
```

$$x \to 4\,e^{(-2\,x)}$$

```
> (D@@5)(f) ; (D@@3)((D@@2)(f)) ;
```

$$x \to -32\,e^{(-2\,x)}$$

$$x \to -32\,e^{(-2\,x)}$$

Diverses représentations graphiques

```
> plot(f, -5..5) ;
```




```
> plot({f, f1 }, -1..5, -2..2) ;
```




```
> plot({f, f1, f2 }, -1..5, -2..2) ;
```




```
> f := x->sin(x) ; d1 := x->x ; d2 := x->x-x^3/6 ;
```

$$f := \sin$$

$$d1 := x \to x$$

$$d2 := x \to x - \frac{1}{6}x^3$$

```
> d3 := x->x-x^3/6+x^5/120 ;
```

$$d3 := x \to x - \frac{1}{6}x^3 + \frac{1}{120}x^5$$

```
> d4 :=  x->x-x^3/6+x^5/120-x^7/720 ;
```

$$d4 := x \to x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{720}x^7$$

```
> plot({f,d1,d2,d3,d4},-2*Pi..2*Pi,-2..2) ;
```



Composition

```
> f := sin ; p := x -> 3*x + 1 ;
```

$$f := \sin$$

$$p := x \to 3\,x + 1$$

```
> f@p ; (f@p)(x) ;
```

$$\sin@p$$

$$\sin(3\,x + 1)$$

```
> (p@@3) ; (p@@3)(x);
```

$$p^{(3)}$$

$$27\,x + 13$$

Dérivation

```
> f := x->x^4-4*x^2 ; g:=D(f) ; h := D(g) ; h := (D@D)(f) ;
```

$$f := x \to x^4 - 4\,x^2$$

$$g := x \to 4\,x^3 - 8\,x$$

$$h := x \to 12\,x^2 - 8$$

$$h := x \to 12\,x^2 - 8$$

```
> plot({f, g}, -3..3,-5..5) ;
```




```
> i := (D @@ 3)(f) ;
```

$$i := x \to 24\,x$$

```
> f:=x->ln(x) ;
```

$$f := \ln$$

```
> seq((D @@ n)(f), n=0..5);
```

$$\ln, a \to \frac{1}{a}, a \to -\frac{1}{a^2}, a \to \frac{2}{a^3}, a \to -\frac{6}{a^4}, a \to \frac{24}{a^5}$$

```
> p := 6 ; j:=(D @@ p)(f) ;
```

$$p := 6$$

$$j := a \to -\frac{120}{a^6}$$

Une primitive définie par une intégrale

```
> f := ln ; F:=x->integrate(f(t),t = 0.. x) ;
```

$$f := \ln$$

$$F := x \to \text{integrate}(\text{f}(t), t = 0 .. x)$$

```
> F(1) ; F(x) ; F(t) ;
```

$$-1$$

$$x \ln(x) - x$$

```
Error, (in limit) invalid arguments
```

```
> int(sin(x),x);
```

$$-\cos(x)$$

```
> F:= unapply(int(f(x),x),x)   ; F(x) ; F(1) ;
```

$$F := x \to x \ln(x) - x$$

$$x \ln(x) - x$$

$$-1$$

```
> plot(F, -1..5, -2..2) ;
```



Une fonction définie par intervalles

```
> f:=proc(x) local Resultat;
> if x<-1 then Resultat:=x^2 elif x<=1 then Resultat:=-2 else
  Resultat:=1/x fi;
> RETURN(Resultat);
> end;
```

$f :=$ **proc**(x)

local *Resultat*;

if $x < -1$ **then** *Resultat* $:= x$^2 **elif** $x \leq 1$ **then** *Resultat* $:= -2$ **else** *Resultat* $:= 1 / x$ **fi**;

RETURN(*Resultat*)

end

```
> map(f,[-10,-3,-1.01,-1,0.8,1,1.01,2,10]);
```

$$\left[100, 9, 1.0201, -2, -2, -2, .9900990099, \frac{1}{2}, \frac{1}{10}\right]$$

```
> L:=[-10,-3,-1.01,-1,0.8,1,1.01,2,10];convert(map(f,L),fraction);
```

$$L := [-10, -3, -1.01, -1, .8, 1, 1.01, 2, 10]$$

$$\left[100, 9, \frac{10201}{10000}, -2, -2, -2, \frac{100}{101}, \frac{1}{2}, \frac{1}{10}\right]$$

```
> plot(f,-4..4,-3..10);
```



Pourquoi se fatiguer : Maple sait le faire !

```
> g := x->piecewise(x<-1,x^2,x<=1,-2,1/x);
```

$$g := x \rightarrow \text{piecewise}\left(x < -1, x^2, x \leq 1, -2, \frac{1}{x}\right)$$

```
> g(x);
```

$$\begin{cases} x^2 & x < -1 \\ -2 & x \leq 1 \\ \dfrac{1}{x} & \textit{otherwise} \end{cases}$$

```
> plot(g,-4..4,-3..10);
```


10
8
6
4
2
-4
-2
0
2
4
-2

[>

Intégration

Groupe calcul formel - IREM de Grenoble

Fiche n° 12

```
> restart;
> f:= x->ln(x) ;
```

$$f := \ln$$

```
> int(f(x), x) ;
```

$$x\ln(x) - x$$

```
> diff(",x) ;
```

$$\ln(x)$$

```
> G:=x->int(sin(x)^2,x);G(x);
```

$$G := x \to \int \sin(x)^2\, dx$$

$$-\frac{1}{2}\cos(x)\sin(x) + \frac{1}{2}x$$

```
> g1 := D(G) ;
```

$$g1 := x \to \sin(x)^2$$

```
> simplify(g1(x)); combine(g1(x), trig) ;
```

$$1 - \cos(x)^2$$

$$\frac{1}{2} - \frac{1}{2}\cos(2x)$$

```
> F := x -> int(cos(t)^4, t=0..x) ;
```

$$F := x \to \int_0^x \cos(t)^4\, dt$$

```
> D(F) ;
```

$$x \to \cos(x)^4$$

```
> F(x);
```

$$\frac{1}{4}\cos(x)^3\sin(x) + \frac{3}{8}\cos(x)\sin(x) + \frac{3}{8}x$$

```
> combine(F(x), trig) ;
```

$$\frac{1}{32}\sin(4x) + \frac{1}{4}\sin(2x) + \frac{3}{8}x$$

```
> g := x -> sqrt(1+x) ;
```

$$g := x \to \sqrt{x+1}$$

```
> G := x -> int(g(t) ,t=0..x) ;
```

$$G := x \to \int_0^x g(t)\, dt$$

```
> D(G)(x) ;
```

$$\sqrt{x+1}$$

```
> D(G) ; G(0) ; G(x) ;
```

$$g$$

$$0$$

$$\frac{2}{3}(x+1)^{3/2}-\frac{2}{3}$$

```
>
```

Nombres complexes

Groupe calcul formel - IREM de Grenoble

Fiche n° 13

```
> restart;
```

Forme algébrique et trigonométrique

```
> polar(-1) ;
```

$$\mathrm{polar}(-1)$$

```
> convert(I, polar) ;
```

$$\mathrm{polar}\left(1, \frac{1}{2}\pi\right)$$

```
> convert(1+I, polar) ;
```

$$\mathrm{polar}\left(\sqrt{2}, \frac{1}{4}\pi\right)$$

```
> evalc((1+I)^5) ;
```

$$-4-4I$$

Calculs, résolutions d'équations

```
> J := exp(I*2*Pi/3) ;
```

$$J := -\frac{1}{2}+\frac{1}{2}I\sqrt{3}$$

```
> evalc(1+J+J^2) ;
```

$$0$$

```
> polar(J) ;
```

$$\mathrm{polar}\left(1, \frac{2}{3}\pi\right)$$

```
> abs(3+5*I) ;
```

$$\sqrt{34}$$

```
> evalc(Re(E^(I*Pi/4))) ;
```

$$e^{(-1/4\,\pi^2\,(1/2-1/2\,\mathrm{signum}(E)))}\cos\left(\frac{1}{4}\pi\ln(|E|)\right)$$

```
> Re(1+I) ; Im((3-I)/(7+2*I)) ;
```

$$1$$

$$\frac{-13}{53}$$

```
> z := polar(2, Pi/4) ;
```

$$z := \mathrm{polar}\left(2, \frac{1}{4}\pi\right)$$

```
> evalc(z) ;
```

$$\sqrt{2}+I\sqrt{2}$$

```
> polar(z^3) ;
```

$$\mathrm{polar}\left(8, \mathrm{argument}\left(\mathrm{polar}\left(2, \frac{1}{4}\pi\right)^3\right)\right)$$

```
> argument(z) ;
```

$$\frac{1}{4}\pi$$

```
> evalc(argument(z^3)) ;
```

$$\frac{3}{4}\pi$$

```
> conjugate(2+3*I) ; evalc(conjugate(E^(I*Pi/2))) ;
```

$$2-3I$$

$$e^{(-1/2\,\pi^2\,(1/2-1/2\,\mathrm{signum}(E)))}\cos\left(\frac{1}{2}\pi\ln(|E|)\right)-I\,e^{(-1/2\,\pi^2\,(1/2-1/2\,\mathrm{signum}(E)))}\sin\left(\frac{1}{2}\pi\ln(|E|)\right)$$

```
> solve(z^2+z+1=0,z) ;
```

$$-\frac{1}{2}+\frac{1}{2}I\sqrt{3}, -\frac{1}{2}-\frac{1}{2}I\sqrt{3}$$

```
> solve(z^17-1=0,z) ;
```

$$1, \cos\left(\frac{2}{17}\pi\right)+I\sin\left(\frac{2}{17}\pi\right), \cos\left(\frac{4}{17}\pi\right)+I\sin\left(\frac{4}{17}\pi\right), \cos\left(\frac{6}{17}\pi\right)+I\sin\left(\frac{6}{17}\pi\right),$$

$$\cos\left(\frac{8}{17}\pi\right)+I\sin\left(\frac{8}{17}\pi\right), -\cos\left(\frac{7}{17}\pi\right)+I\sin\left(\frac{7}{17}\pi\right), -\cos\left(\frac{5}{17}\pi\right)+I\sin\left(\frac{5}{17}\pi\right),$$

$$-\cos\left(\frac{3}{17}\pi\right)+I\sin\left(\frac{3}{17}\pi\right), -\cos\left(\frac{1}{17}\pi\right)+I\sin\left(\frac{1}{17}\pi\right), -\cos\left(\frac{1}{17}\pi\right)-I\sin\left(\frac{1}{17}\pi\right),$$

$$-\cos\left(\frac{3}{17}\pi\right)-I\sin\left(\frac{3}{17}\pi\right), -\cos\left(\frac{5}{17}\pi\right)-I\sin\left(\frac{5}{17}\pi\right), -\cos\left(\frac{7}{17}\pi\right)-I\sin\left(\frac{7}{17}\pi\right),$$

$$\cos\left(\frac{8}{17}\pi\right)-I\sin\left(\frac{8}{17}\pi\right), \cos\left(\frac{6}{17}\pi\right)-I\sin\left(\frac{6}{17}\pi\right), \cos\left(\frac{4}{17}\pi\right)-I\sin\left(\frac{4}{17}\pi\right),$$

$$\cos\left(\frac{2}{17}\pi\right)-I\sin\left(\frac{2}{17}\pi\right)$$

```
> f := z -> (2*z +I) / (I*z +1) ;
```

$$f := z \to \frac{2z+I}{Iz+1}$$

```
> f(1) ;
```

$$\frac{3}{2}-\frac{1}{2}I$$

```
> z:=a+I*b ;simplify(evalc( Re(f(z)))) ;
```

$$z := a+Ib$$

$$3\,\frac{a}{b^2-2b+1+a^2}$$

```
> z := 'z' :  solve((4-I-z)/(-I-z)=(3-sqrt(3))/2*(1+I), z) ;
```

$$-\frac{-5-I+I\sqrt{3}-\sqrt{3}}{-1+\sqrt{3}-3I+I\sqrt{3}}$$

```
> a := 4-I ; b := 1+sqrt(3)+I*(2+sqrt(3)) ; d := -I ;
```

$$a := 4-I$$

$$b := 1+\sqrt{3}+I(2+\sqrt{3})$$

$$d := -I$$

```
> Z :=(a-b)/(d-b) ;
```

$$Z := \frac{3-I-\sqrt{3}-I(2+\sqrt{3})}{-I-1-\sqrt{3}-I(2+\sqrt{3})}$$

```
> Z :=factor(Z) ;
```

$$Z := -\frac{1}{2}I\sqrt{3}+\frac{3}{2}I+\frac{3}{2}-\frac{1}{2}\sqrt{3}$$

```
> Re(Z) ; Im(Z) ;
```

$$\frac{3}{2}-\frac{1}{2}\sqrt{3}$$

$$\frac{3}{2}-\frac{1}{2}\sqrt{3}$$

```
>
```

Equations différentielles

Groupe calcul formel - IREM de Grenoble

Fiche n° 14

```
> restart;
```

équations du premier ordre

```
> equa:= diff(y(x),x)-2*y(x)=0;
```

$$equa := \left(\frac{\partial}{\partial x}\mathrm{y}(x)\right) - 2\,\mathrm{y}(x) = 0$$

```
> dsolve(equa,y(x));
```

$$\mathrm{y}(x) = \mathbf{e}^{(2\,x)}\,\_C1$$

avec une condition initiale

```
> dsolve({equa,y(0)=1},y(x));
```

$$\mathrm{y}(x) = \mathbf{e}^{(2\,x)}$$

équations du second ordre

```
> equa1:=diff(y(x),x$2)-6*diff(y(x),x)+9*y(x)=0;
```

$$equa1 := \left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) - 6\left(\frac{\partial}{\partial x}\mathrm{y}(x)\right) + 9\,\mathrm{y}(x) = 0$$

```
> dsolve(equa1,y(x));
```

$$\mathrm{y}(x) = \_C1\,\mathbf{e}^{(3\,x)} + \_C2\,\mathbf{e}^{(3\,x)}\,x$$

avec conditions initiales

```
> dsolve({equa,y(0)=1,D(y)(0)=2},y(x));
```

$$\mathrm{y}(x) = \mathbf{e}^{(2\,x)}$$

à coefficients non constants

```
> equa2:= t*diff(x(t),t)-2*x(t)=t^3*exp(t);
```

$$equa2 := t\left(\frac{\partial}{\partial t}\mathrm{x}(t)\right) - 2\,\mathrm{x}(t) = t^3\,\mathbf{e}^t$$

```
> dsolve(equa2,x(t));
```

$$\mathrm{x}(t) = t^2\,\mathbf{e}^t + t^2\,\_C1$$

Maple peut résoudre les systèmes d'équations différentielles

```
> sys:={diff(x(t),t)=-4*x(t)+2*y(t)+z(t),diff(y(t),t)=-11*x(t)+6*y
  (t)+2*z(t),diff(z(t),t)=3*x(t)-3*y(t)+z(t)};
```

$$sys := \{\frac{\partial}{\partial t}\mathrm{x}(t) = -4\,\mathrm{x}(t) + 2\,\mathrm{y}(t) + \mathrm{z}(t), \frac{\partial}{\partial t}\mathrm{y}(t) = -11\,\mathrm{x}(t) + 6\,\mathrm{y}(t) + 2\,\mathrm{z}(t),$$

$$\frac{\partial}{\partial t}\mathrm{z}(t) = 3\,\mathrm{x}(t) - 3\,\mathrm{y}(t) + \mathrm{z}(t)\}$$

```
> dsolve(sys,{x(t),y(t),z(t)});
```

$$\{\mathrm{z}(t) = \_C1\,\mathbf{e}^t - \frac{3}{2}\_C1\,t^2\,\mathbf{e}^t + 3\,\_C2\,t\,\mathbf{e}^t + 9\,\_C2\,t^2\,\mathbf{e}^t - \frac{9}{2}\_C3\,t^2\,\mathbf{e}^t - 3\,\_C3\,t\,\mathbf{e}^t,$$

$$x(t) = \_C1\, t\, e^t - \frac{1}{2}\_C1\, t^2\, e^t + \_C2\, e^t + 3\,\_C2\, t^2\, e^t - 5\,\_C2\, t\, e^t - \frac{3}{2}\_C3\, t^2\, e^t + 2\,\_C3\, t\, e^t,$$

$$y(t) = 2\,\_C1\, t\, e^t - \frac{1}{2}\_C1\, t^2\, e^t + 3\,\_C2\, t^2\, e^t - 11\,\_C2\, t\, e^t + \_C3\, e^t - \frac{3}{2}\_C3\, t^2\, e^t + 5\,\_C3\, t\, e^t\}$$

on peut utiliser la transformation de Laplace

```
> del := diff(y(t),t$2) + 5*diff(y(t),t) + 6*y(t) = 0;
> dsolve({del, y(0)=0, D(y)(0)=1}, y(t),method=laplace);
```

$$del1 := \left(\frac{\partial^2}{\partial t^2}\mathrm{y}(t)\right) + 5\left(\frac{\partial}{\partial t}\mathrm{y}(t)\right) + 6\,\mathrm{y}(t) = 0$$

$$\mathrm{y}(t) = -e^{(-3\,t)} + e^{(-2\,t)}$$

on peut obtenir une approximation numérique de la solution

```
> equa3:=diff(y(x),x)=x*y(x);
```

$$equa3 := \frac{\partial}{\partial x}\mathrm{y}(x) = x\,\mathrm{y}(x)$$

```
> f:=dsolve({equa3,y(0)=1},y(x),numeric);
```

$f :=$ **proc**($rkf45\_x$) ... **end**

```
> f(1);
```

$$[x = 1, \mathrm{y}(x) = 1.648721368254647]$$

on peut alors représenter la solution à l'aide de l'outil odeplot

```
> with(plots):odeplot(f,[x,y(x)],0..1,color=black);
  # ATTENTION : cette méthode nécessite les conditions initiales
  au même point
```



Un problème de BTS

```
> restart;
```

On considère l'équation différentielle suivante... Imaginer la suite de l'énoncé.

```
> eqd1:=4*(D@@2)(y)(x)-4*D(y)(x)+y(x)=8*exp(x/2);
```

$$eqd1 := 4\,(D^{(2)})(y)(x) - 4\,\mathrm{D}(y)(x) + \mathrm{y}(x) = 8\,\mathrm{e}^{(1/2\,x)}$$

```
> membre1:=op(1,eqd1);membre2:=op(2,eqd1);
```

$$membre1 := 4\,(D^{(2)})(y)(x) - 4\,\mathrm{D}(y)(x) + \mathrm{y}(x)$$

$$membre2 := 8\,\mathrm{e}^{(1/2\,x)}$$

```
> eqd0:=membre1=0;
```

$$eqd0 := 4\,(D^{(2)})(y)(x) - 4\,\mathrm{D}(y)(x) + \mathrm{y}(x) = 0$$

```
> dsolve(eqd0,y(x));
```

$$\mathrm{y}(x) = \_C1\,\mathrm{e}^{(1/2\,x)} + \_C2\,\mathrm{e}^{(1/2\,x)}\,x$$

```
> g:=x->x^2*exp(x/2);
```

$$g := x \to x^2\,\mathrm{e}^{(1/2\,x)}$$

```
> subs(y=g,membre1);
```

$$4\,(D^{(2)})(g)(x) - 4\,\mathrm{D}(g)(x) + \mathrm{g}(x)$$

```
> simplify(");membre2;
```

$$8\,\mathrm{e}^{(1/2\,x)}$$

$$8\,\mathrm{e}^{(1/2\,x)}$$

```
> dsolve({eqd1},y(x));
```

$$\mathrm{y}(x) = x^2\,\mathrm{e}^{(1/2\,x)} + \_C1\,\mathrm{e}^{(1/2\,x)} + \_C2\,\mathrm{e}^{(1/2\,x)}\,x$$

```
> cond:=y(0)=-2,(D@@1)(y)(0)=-1:eqd:={eqd1,cond}:sol:=dsolve(eqd,y
(x));
```

$$sol := \mathrm{y}(x) = x^2\,\mathrm{e}^{(1/2\,x)} - 2\,\mathrm{e}^{(1/2\,x)}$$

```
> f:=unapply(op(2,sol),x);
```

$$f := x \to x^2\,\mathrm{e}^{(1/2\,x)} - 2\,\mathrm{e}^{(1/2\,x)}$$

```
> plot(f,-6..2,color=black);
```




```
> with(plots):dsolve(eqd,y(x),numeric):odeplot(",x,-2..2,color=bla
  ck);
  # plus directement, mais on sait que ça ne marche pas si les
  conditions initiales ne sont pas données au même point.
```




```
> solve(f(x)=0,x);
```

$$\sqrt{2}, -\sqrt{2}$$

```
> f1:=D(f);
```

$$f1 := x \to 2\,x\,e^{(1/2\,x)} + \frac{1}{2}x^2\,e^{(1/2\,x)} - e^{(1/2\,x)}$$

```
> sol:=solve({f1(x)=0},x):
  for i from 1 to nops({sol}) do x.i:=subs(sol[i],x) od;
  evalf(seq(f(x.i),i=1..nops({sol})));
```

$$x1 := -2 + \sqrt{6}$$

$$x2 := -2 - \sqrt{6}$$

-2.251050482, 1.923870254

```
> int(f(x),x);
```

$$2\,x^2\,e^{(1/2\,x)} - 8\,e^{(1/2\,x)}\,x + 12\,e^{(1/2\,x)}$$

```
> int(f(x),x=-sqrt(2)..sqrt(2));
```

$$16\,e^{(1/2\sqrt{2})} - 8\sqrt{2}\,e^{(1/2\sqrt{2})} + (-16 - 8\sqrt{2})\,e^{(-1/2\sqrt{2})}$$

```
> aire:=evalf(abs("),4);
```

$$aire := 3.96$$

```
>
```

Quelques suites classiques

Groupe calcul formel - IREM de Grenoble

Fiche n° 15

```
> restart;
```

Programmation de suites récurrentes : Fibonacci et suite tchèque

```
> fibo := proc(u0, u1, n) options remember ;
> if n=0 then u0 elif n=1 then u1
> else fibo(u0, u1, n-1) + fibo(u0, u1, n-2)fi
> end ;
```

fibo := **proc**(*u0*, *u1*, *n*)
option *remember*;
if $n = 0$ **then** *u0* **elif** $n = 1$ **then** *u1* **else** fibo(*u0*, *u1*, $n - 1$) + fibo(*u0*, *u1*, $n - 2$) **fi**
end

```
> fibo(1, 1 , 25);
```

$$121393$$

```
> seq(fibo(1,1,i), i=0..20) ;
```

1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, 1597, 2584, 4181, 6765, 10946

```
> nbcfib := n -> 1+trunc(evalf(log10(fibo(1,1,n)))) ;
```

$$nbcfib := n \rightarrow 1 + \mathrm{trunc}(\mathrm{evalf}(\log 10(\mathrm{fibo}(1, 1, n))))$$

```
> nbcfib(25) ;
```

$$6$$

```
> nbcfib(50) ;
```

$$11$$

```
> fibo(1,1,50) ;
```

$$20365011074$$

```
> tchq := proc(u0, n) options remember ;
> if n=0 then u0 elif type(tchq(u0, n-1), even) then iquo(tchq
> (u0, n-1), 2) else 3* tchq(u0, n-1) + 1 fi ;
> end ;
```

tchq := **proc**(*u0*, *n*)
option *remember*;
if $n = 0$ **then** *u0*
elif type(tchq(*u0*, $n - 1$), *even*) **then** iquo(tchq(*u0*, $n - 1$), 2)
else 3*tchq(*u0*, $n - 1$) + 1
fi
end

```
> seq(tchq(27, p), p=0..150) ;
```

27, 82, 41, 124, 62, 31, 94, 47, 142, 71, 214, 107, 322, 161, 484, 242, 121, 364, 182, 91, 274, 137, 412, 206, 103, 310, 155, 466, 233, 700, 350, 175, 526, 263, 790, 395, 1186, 593, 1780, 890, 445, 1336, 668, 334, 167, 502, 251, 754 277, 1132, 566, 283, 850, 425, 1276, 638, 319, 958, 479, 1438, 719, 2158, 1079, 3238, 1619, 4858, 2429, 7288, 3644, 1822, 911, 2734, 1367, 4102, 2051, 6154, 3077, 9232, 4616, 2308, 1154, 577, 1732, 866, 433, 1300, 650, 325, 976, 488, 244, 122, 61, 184, 92, 46, 23, 70, 35, 106, 53, 160, 80, 40, 20, 10, 5, 16, 8, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1, 4, 2, 1

```
> seq2_tchq := proc(u0) local l, u, n :
> l := [] : u := u0 : n := 0 :
> while u<>1 do l := [op(l), u] : n := n+1 : u := tchq(u0, n)  od
  :
> l := [op(l), u] :
> RETURN(l) :
> end ;
```

seq2_tchq := **proc**(*u0*)
local *l*, *u*, *n*;
$l := [\ ]$;
$u := u0$;
$n := 0$;
while $u \neq 1$ **do** $l := [\mathrm{op}(l), u]$; $n := n + 1$; $u := \mathrm{tchq}(u0, n)$ **od**;
$l := [\mathrm{op}(l), u]$;
RETURN(l)
end

```
> seq2_tchq(27) ;
```

[27, 82, 41, 124, 62, 31, 94, 47, 142, 71, 214, 107, 322, 161, 484, 242, 121, 364, 182, 91, 274, 137, 412, 206, 103, 310, 155, 466, 233, 700, 350, 175, 526, 263, 790, 395, 1186, 593, 1780, 890, 445, 1336, 668, 334, 167, 502, 251, 754, 377, 1132, 566, 283, 850, 425, 1276, 638, 319, 958, 479, 1438, 719, 2158, 1079, 3238, 1619, 4858, 2429, 7288, 3644, 1822, 911, 2734, 1367, 4102, 2051, 6154, 3077, 9232, 4616, 2308, 1154, 577, 1732, 866, 433, 1300, 650, 325, 976, 488, 244, 122, 61, 184, 92, 46, 23, 70, 35, 106, 53, 160, 80, 40, 20, 10, 5, 16, 8, 4, 2, 1]

```
>
```

Combinatoire avec programmation

Groupe calcul formel - IREM de Grenoble

Fiche n° 16

```
> restart;
```

Programmation itérative et récursive des C(n, p)

```
> n:=20 ; seq(binomial(n,p),p=0..n);
```

$$n := 20$$

1, 20, 190, 1140, 4845, 15504, 38760, 77520, 125970, 167960, 184756, 167960, 125970, 77520, 38760, 15504, 4845, 1140, 190, 20, 1

```
> bin1 := proc(k,l)
> k!/(l!*(k-l)!);
> end ;
```

bin1 := **proc**(*k*, *l*) *k*! / (*l*!*(*k* − *l*)!) **end**

```
> seq(bin1(n,p),p=0..n);
```

1, 20, 190, 1140, 4845, 15504, 38760, 77520, 125970, 167960, 184756, 167960, 125970, 77520, 38760, 15504, 4845, 1140, 190, 20, 1

```
> bin2:=proc(k,l) options remember ;
> if (l=0)or(k=l) then 1 else bin2(k-1,l) + bin2(k-1,l-1) fi ;
> end ;
```

bin2 := **proc**(*k*, *l*)
option *remember*;
if *l* = 0 **or** *k* = *l* **then** 1 **else** bin2(*k* − 1, *l*) + bin2(*k* − 1, *l* − 1) **fi**
end

```
> seq(bin2(n,p),p=0..n);
```

1, 20, 190, 1140, 4845, 15504, 38760, 77520, 125970, 167960, 184756, 167960, 125970, 77520, 38760, 15504, 4845, 1140, 190, 20, 1

```
> bin3 := proc(k,l) options remember ;
> if l=0 then 1 else iquo(k*bin3(k-1, l-1), l) fi ;
> end ;
```

bin3 := **proc**(*k*, *l*) **option** *remember*; **if** *l* = 0 **then** 1 **else** iquo(*k**bin3(*k* − 1, *l* − 1), *l*) **fi** **end**

```
> seq(bin3(n,p),p=0..n);
```

1, 20, 190, 1140, 4845, 15504, 38760, 77520, 125970, 167960, 184756, 167960, 125970, 77520, 38760, 15504, 4845, 1140, 190, 20, 1

```
> bin4 := proc(k, l) local i, r ;
> r := 1 :
> for i from 1 to l do r := iquo(r*(k-i+1), i) od ;
> r;
> end ;
```

bin4 := **proc**(*k*, *l*) **local** *i*, *r*; *r* := 1; **for** *i* **to** *l* **do** *r* := iquo(*r**(*k* − *i* + 1), *i*) **od**; *r* **end**

```
> bin4(10, 5) ;
```

$$252$$

```
> binomial(10,5) ;
```

$$252$$

```
>
```

Autres suites récurrentes

Groupe calcul formel - IREM de Grenoble

Fiche n° 17

```
> restart;
```

Programmation récursive

```
> s := proc(s0, f, n) options remember ;
> if n = 0 then s0 else f(s(s0, f, n-1)) fi
> end:
> f := x->sqrt((1+x)/2) ;
```

$$f := x \to \sqrt{\frac{1}{2} + \frac{1}{2}x}$$

```
> u := n->s(2,f,n);
```

$$u := n \to \mathrm{s}(2, f, n)$$

```
> u(5);
```

$$\frac{1}{2}\sqrt{2+\sqrt{2+\sqrt{2+\sqrt{2+\sqrt{6}}}}}$$

```
> evalf(u(5));
```

$$1.000846984$$

```
> solve(f(x)=x) ;
```

$$1$$

```
> plot({f(x),x}, x=-1..2);
```




```
> seq(evalf(u(n)), n=0..10) ;
```

2., 1.224744872, 1.054690683, 1.013580456, 1.003389370, 1.000846984, 1.000211724, 1.000052930, 1.000013233, 1.000003308, 1.000000827

La procédure qui suit produit une liste de coordonnées permettant de construire des segments de droite à l'aide de la commande plot.

```
> liste_seg := proc(su, nb) local v1, v2 ;
> if nb = 0 then [[su(0),0]] else
> v1 := evalf(su(nb-1)) : v2 := evalf(su(nb)):
> [op(liste_seg(su, nb-1)), [v1, v2], [v2, v2]]
> fi
> end :
> l:=liste_seg(u,10);
```

$l := [[2, 0], [2., 1.224744872], [1.224744872, 1.224744872], [1.224744872, 1.054690683], [1.054690683, 1.054690683], [1.054690683, 1.013580456], [1.013580456, 1.013580456], [1.013580456, 1.003389370], [1.003389370, 1.003389370], [1.003389370, 1.000846984], [1.000846984, 1.000846984], [1.000846984, 1.000211724], [1.000211724, 1.000211724], [1.000211724, 1.000052930], [1.000052930, 1.000052930], [1.000052930, 1.000013233], [1.000013233, 1.000013233], [1.000013233, 1.000003308], [1.000003308, 1.000003308], [1.000003308, 1.000000827], [1.000000827, 1.000000827]]$

```
> plot({f(x),x,l},x=0..2,y=0..2) ;
```




```
> f := x->cos(x) ;
```

$$f := \cos$$

```
> u := n->s(0.5,f,n);
```

$$u := n \rightarrow s(.5, f, n)$$

```
> l:=liste_seg(u,10):
> plot({f(x),x,l},x=-1..2,y=0..1) :
```



1
0.8
0.6
y
0.4
0.2
-1
-0.5
0
0.5
1
1.5
2
x

[>

Courbes paramétriques

Groupe calcul formel - IREM de Grenoble

Fiche n° 18

```
> restart;
```

Cissoïde de Dioclès

```
> x:= t-> a*t^2/(1+t^2) ; y := t -> t*x(t) ;
```

$$x := t \to \frac{a\,t^2}{1+t^2}$$

$$y := t \to t\,\mathrm{x}(t)$$

```
> a:=1 ; plot([x(t), y(t),t=-100..100], x=0..2, y=-8..8) ;
```

$$a := 1$$



Strophoïde droite

```
> x:= t-> a*(t^2-1)/(1+t^2) ; y := t -> t*x(t) ;
```

$$x := t \to \frac{a\,(t^2-1)}{1+t^2}$$

$$y := t \to t\,\mathrm{x}(t)$$

```
> a:= 1 ; plot([x(t), y(t),t=-100..100], x=-1..1, y = -1..1) ;
```

$$a := 1$$



Folium de Descartes

```
> x:= t-> a*2*t/(1+t^3) ; y := t -> t*x(t) ;
```

$$x := t \to 2\,\frac{a\,t}{1+t^3}$$

$$y := t \to t\,\mathrm{x}(t)$$

```
> a:= 10; plot([x(t), y(t),t=-10..10], x=-20..20, y=-20..20) ;
```

$$a := 10$$



Courbes de Lissajous

```
> a := 2 ; b := 3 ; x:= t-> sin(a*t) ; y := t -> sin(b*t) ;
```

$$a := 2$$

$$b := 3$$

$$x := t \rightarrow \sin(a\,t)$$

$$y := t \rightarrow \sin(b\,t)$$

```
> plot([x(t), y(t),t=-Pi..Pi], x=-1..1, y=-1..1) ;
```




```
> f := t -> cos(2*t) ;
```

$$f := t \rightarrow \cos(2\,t)$$

```
> g := t -> sin(3*t) ;
```

$$g := t \rightarrow \sin(3\,t)$$

```
> plot({[f(t), g(t), t = 0..2*Pi]}, x=-1..1, y=-1..1) ;
```




```
> limit((g(t)+1)/(t-Pi/2)^2, t=Pi/2) ;
```

$$\frac{9}{2}$$

```
> limit((f(t)+1)/(t-Pi/2)^2, t=Pi/2) ;
```

$$2$$

```
>
```

Courbes en coordonnées polaires

Groupe calcul formel - IREM de Grenoble

Fiche n° 19

```
> restart;
```

Conchoïdes de cercle...Cardioïde pour a = b

```
> a:=10 : b:=10 : r:=t->a*cos(t)+b ;
```

$$r := t \to a\cos(t) + b$$

```
> plot([r(t),t,t=-Pi..Pi], coords=polar) ;
```




```
> a:=10 : b:=4 : r:=t->a*cos(t)+b ;
```

$$r := t \to a\cos(t) + b$$

```
> plot([r(t), t, t=-Pi..Pi], coords=polar) ;
```



Conchoïdes de droite...

```
> a:=2 : b:=1 : r:=t->a/cos(t)+b ;
```

$$r := t \rightarrow \frac{a}{\cos(t)} + b$$

```
> plot([r(t), t, t=-Pi..Pi],  x=0.9..3.1, y=-50..50, coords=polar)
  ;
```



Les intervalles horizontal et vertical passés en paramètres ne sont pas pris en compte. C'est un bug ! Passons en coordonnées paramétriques.

```
> plot([r(t)*cos(t), r(t)*sin(t), t=-Pi..Pi], x=0.9..3.1,
  y=-50..50);
>
```



Lemniscate de Bernouilli...

```
> a:=10 : r:=t->a*sqrt(cos(2*t)) ;
```

$$r := t \to a\sqrt{\cos(2\,t)}$$

```
> plot([r(t), t, t=-Pi..Pi], coords=polar, numpoints=400, title
  =`lemniscate`) ;
```



Trèfle...

```
> r:=t->abs(sin(2*t)) :
> plot([r(t), t, t=-Pi..Pi], coords=polar, numpoints=100, title =
  `trèfle`) ;
```




```
> r:=t->cos(2*t)/(1+2*sin(t)) ;
```

$$r := t \rightarrow \frac{\cos(2\,t)}{1 + 2\sin(t)}$$

```
> plot([r(t), t, t=-Pi..Pi], coords=polar) ;
```




```
> plot([r(t)*cos(t), r(t)*sin(t), t=-Pi..Pi], x=-3..3,y=-3..3) ;
```



Spirales : Archimède, Fermat, Lituus, Galilée, Poinsot, hyperbolique, log..

```
> a:=1 : r:=t->a*t ; archi:=r :
```

$$r := t \to a\, t$$

```
> plot([r(t), t, t=-Pi..3*Pi], coords=polar) ;
```




```
> a:=1 : r:=t->a*sqrt(t) ; fermat:=r :
```

$$r := t \to a\sqrt{t}$$

```
> plot([r(t), t, t=0..4*Pi], coords=polar) ;
```




```
> a:=10 : r:=t->a / sqrt(t) ; lituus:=r :
```

$$r := t \rightarrow \frac{a}{\sqrt{t}}$$

```
> plot([r(t), t, t=0.01..6*Pi], coords=polar) ;
```




```
> plot([r(t)*cos(t), r(t)*sin(t), t=0.01..6*Pi], x=-10..20) ;
```




```
> galil:=t->1-a*t^2 ; poinsot:=t->1/ch(a*t) ;
```

$$galil := t \rightarrow 1 - a\,t^2$$

$$poinsot := t \rightarrow \frac{1}{\operatorname{ch}(a\,t)}$$

```
> plot([r(t), t, t=0..6*Pi], coords=polar) ;
```




```
> spilog:=t->k^t ; spihyper:=t->a/t ;
```

$$spilog := t \rightarrow k^t$$

$$spihyper := t \rightarrow \frac{a}{t}$$

```
> plot([r(t), t, t=0..6*Pi], coords=polar) ;
```



Rosace

```
> rosace:=t->sin(2*t/3) :  r:=rosace :
> plot([r(t), t, t=0..6*Pi], coords=polar, title=`rosace`) ;
```

rosace




```
> trissec:=t->t/(t-Pi/4) : r:=trissec :
> plot([r(t), t, t=-3*Pi..3*Pi], x=-10..10,coords=polar, title =
  `trissectrice`) ;
```




```
> plot([r(t)*cos(t), r(t)*sin(t), t=-3*Pi..3*Pi],
  x=-5..5,y=-5..5,title = `trissectrice`) ;
```




```
>
```

Matrices

Groupe calcul formel - IREM de Grenoble

Fiche n° 20

```
> restart;
> with(linalg) :
Warning, new definition for norm
Warning, new definition for trace
> A := array([[1,2,3], [3,-1,6], [4,8,12]]) ;
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 3 & -1 & 6 \\ 4 & 8 & 12 \end{bmatrix}$$

```
> det(A) ;
```

$$0$$

```
> pc :=charpoly(A,x) ;
```

$$pc := x^3 - 12\,x^2 - 67\,x$$

```
> subs(x=A, pc) ;
```

$$A^3 - 12\,A^2 - 67\,A$$

```
> evalm(") ;
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

coucou Hamilton Cayley !

```
> factor(pc) ;
```

$$x\,(x^2 - 12\,x - 67)$$

```
> solve(pc=0) ;
```

$$0, 6 + \sqrt{103}, 6 - \sqrt{103}$$

```
> eigenvals(A) ;
```

$$0, 6 + \sqrt{103}, 6 - \sqrt{103}$$

```
> R := randmatrix(4,4) ;
```

$$R := \begin{bmatrix} -37 & -35 & 97 & 50 \\ 79 & 56 & 49 & 63 \\ 57 & -59 & 45 & -8 \\ -93 & 92 & 43 & -62 \end{bmatrix}$$

```
> det(R) ;
```

$$144779126$$

```
> evalm(R^(-1)) ;
```

$$\begin{bmatrix} \frac{-772933}{144779126} & \frac{337229}{72389563} & \frac{995141}{144779126} & \frac{-33201}{72389563} \\ \frac{-217228}{72389563} & \frac{418634}{72389563} & \frac{-258561}{72389563} & \frac{283565}{72389563} \\ \frac{571385}{144779126} & \frac{162229}{72389563} & \frac{1000351}{144779126} & \frac{330704}{72389563} \\ \frac{455503}{72389563} & \frac{227869}{72389563} & \frac{-783131}{72389563} & \frac{-467639}{72389563} \end{bmatrix}$$

```
> t := toeplitz([a,b,c]) ;
```

$$t := \begin{bmatrix} a & b & c \\ b & a & b \\ c & b & a \end{bmatrix}$$

```
> invt := evalm(t^(-1)) ;
```

$$invt := \begin{bmatrix} \frac{a^2-b^2}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} & -\frac{b}{a^2+c\,a-2\,b^2} & -\frac{-b^2+c\,a}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} \\ -\frac{b}{a^2+c\,a-2\,b^2} & \frac{a+c}{a^2+c\,a-2\,b^2} & -\frac{b}{a^2+c\,a-2\,b^2} \\ -\frac{-b^2+c\,a}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} & -\frac{b}{a^2+c\,a-2\,b^2} & \frac{a^2-b^2}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} \end{bmatrix}$$

```
> evalm(t&*invt) ;
```

$$\left[\frac{a\,(a^2-b^2)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} - \frac{b^2}{a^2+c\,a-2\,b^2} - \frac{c\,(-b^2+c\,a)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a},\right.$$

$$-\frac{a\,b}{a^2+c\,a-2\,b^2} + \frac{b\,(a+c)}{a^2+c\,a-2\,b^2} - \frac{c\,b}{a^2+c\,a-2\,b^2},$$

$$\left.-\frac{a\,(-b^2+c\,a)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} - \frac{b^2}{a^2+c\,a-2\,b^2} + \frac{c\,(a^2-b^2)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a}\right]$$

$$\left[\frac{b\,(a^2-b^2)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} - \frac{a\,b}{a^2+c\,a-2\,b^2} - \frac{b\,(-b^2+c\,a)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a},\right.$$

$$-2\,\frac{b^2}{a^2+c\,a-2\,b^2} + \frac{a\,(a+c)}{a^2+c\,a-2\,b^2},$$

$$\left.\frac{b\,(a^2-b^2)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} - \frac{a\,b}{a^2+c\,a-2\,b^2} - \frac{b\,(-b^2+c\,a)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a}\right]$$

$$\left[-\frac{a\,(-b^2+c\,a)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} - \frac{b^2}{a^2+c\,a-2\,b^2} + \frac{c\,(a^2-b^2)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a},\right.$$

$$-\frac{a\,b}{a^2+c\,a-2\,b^2} + \frac{b\,(a+c)}{a^2+c\,a-2\,b^2} - \frac{c\,b}{a^2+c\,a-2\,b^2},$$

$$\left.\frac{a\,(a^2-b^2)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a} - \frac{b^2}{a^2+c\,a-2\,b^2} - \frac{c\,(-b^2+c\,a)}{a^3-2\,a\,b^2+2\,c\,b^2-c^2\,a}\right]$$

```
>
```

Développements limités

Groupe calcul formel - IREM de Grenoble

Fiche n° 21

```
> restart;
```

Quelques exemples issus de colle de sup...

```
> taylor((1+x)^(1/x), x=0, 5);
```

$$e-\frac{1}{2}e\,x+\frac{11}{24}e\,x^2-\frac{7}{16}e\,x^3+O(x^4)$$

```
> taylor(x^2/(exp(x)-1), x=0, 4);
```

$$x-\frac{1}{2}x^2+O(x^3)$$

```
> f:= x->exp(x)-sqrt(1+2*x);taylor(f(x), x=0, 7);
```

$$f:=x\to e^x-\sqrt{1+2\,x}$$

$$x^2-\frac{1}{3}x^3+\frac{2}{3}x^4-\frac{13}{15}x^5+\frac{473}{360}x^6+O(x^7)$$

```
> f:=x->(cos(x))^sin(x);taylor(f(x), x=0, 5);
```

$$f:=x\to\cos(x)^{\sin(x)}$$

$$1-\frac{1}{2}x^3+O(x^5)$$

```
> f:=x->ln(x)/x^2 ;taylor(f(x), x=1, 5);
```

$$f:=x\to\frac{\ln(x)}{x^2}$$

$$x-1-\frac{5}{2}(x-1)^2+\frac{13}{3}(x-1)^3-\frac{77}{12}(x-1)^4+O((x-1)^5)$$

```
> f:=x->sqrt(1-sqrt(1-x^2)) ;taylor(f(x), x=0, 7);
```

$$f:=x\to\sqrt{1-\sqrt{1-x^2}}$$

$$\frac{1}{2}\sqrt{2}\,x+\frac{1}{16}\sqrt{2}\,x^3+\frac{7}{256}\sqrt{2}\,x^5+\frac{33}{2048}\sqrt{2}\,x^7+O(x^9)$$

```
> f:=x->ln(3*exp(x)+exp(-x)) ;taylor(f(x), x=0, 5);
```

$$f:=x\to\ln(3\,e^x+e^{(-x)})$$

$$\ln(4)+\frac{1}{2}x+\frac{3}{8}x^2-\frac{1}{8}x^3-\frac{1}{64}x^4+O(x^5)$$

```
> f:=x->ln(sqrt(1+x)+sqrt(1-x)) ;taylor(f(x), x=0, 5);
```

$$f:=x\to\ln(\sqrt{1+x}+\sqrt{1-x})$$

$$\ln(2)-\frac{1}{8}x^2-\frac{3}{64}x^4+O(x^5)$$

```
> f:=x->ln(ln(E+x)) ;taylor(f(x), x=0, 4);
```

$$f := x \to \ln(\ln(E+x))$$

$$\ln(\ln(E)) + \frac{1}{E\ln(E)}x + \left(-\frac{1}{2}\frac{1}{E^2\ln(E)} - \frac{1}{2}\frac{1}{E^2\ln(E)^2}\right)x^2 + \left(\frac{1}{3}\frac{1}{E^3\ln(E)} + \frac{1}{6}\frac{1}{E^3\ln(E)^2} + \frac{1}{3}\frac{\ln(E)+1}{E^3\ln(E)^3}\right)x^3 + O(x^4)$$

```
> f:=x->exp(cos(x));taylor(f(x), x=0, 8);
```

$$f := x \to e^{\cos(x)}$$

$$e - \frac{1}{2}e\,x^2 + \frac{1}{6}e\,x^4 - \frac{31}{720}e\,x^6 + O(x^8)$$

```
> f:=x->ln(1+x)/(1+x)  ;taylor(f(x), x=0, 5);
```

$$f := x \to \frac{\ln(1+x)}{1+x}$$

$$x - \frac{3}{2}x^2 + \frac{11}{6}x^3 - \frac{25}{12}x^4 + O(x^5)$$

```
> f:=x->x/(exp(x)-1)  ;taylor(f(x), x=0, 6);
```

$$f := x \to \frac{x}{e^x - 1}$$

$$1 - \frac{1}{2}x + \frac{1}{12}x^2 - \frac{1}{720}x^4 + O(x^5)$$

```
> f:=x->(cos(x))^(1/x^2)  ;taylor(f(x), x=0, 7);
```

$$f := x \to \cos(x)^{\left(\frac{1}{x^2}\right)}$$

$$e^{(-1/2)} - \frac{1}{12}e^{(-1/2)}x^2 - \frac{3}{160}e^{(-1/2)}x^4 + O(x^5)$$

```
> f:=x->(1+sin(x))^cos(x)  ;taylor(f(x), x=0, 5);
```

$$f := x \to (1+\sin(x))^{\cos(x)}$$

$$1 + x - \frac{2}{3}x^3 - \frac{1}{4}x^4 + O(x^5)$$

```
> f:=x->tan(x)  ;taylor(f(x), x=Pi/4, 4);
```

$$f := \tan$$

$$1 + 2\left(x - \frac{1}{4}\pi\right) + 2\left(x - \frac{1}{4}\pi\right)^2 + \frac{8}{3}\left(x - \frac{1}{4}\pi\right)^3 + O\left(\left(x - \frac{1}{4}\pi\right)^4\right)$$

```
> f:=x->exp(sin(x))  ;taylor(f(x), x=Pi/6, 5);
```

$$f := x \to e^{\sin(x)}$$

$$e^{(1/2)} + \frac{1}{2}e^{(1/2)}\sqrt{3}\left(x - \frac{1}{6}\pi\right) + \frac{1}{8}e^{(1/2)}\left(x - \frac{1}{6}\pi\right)^2 - \frac{7}{48}e^{(1/2)}\sqrt{3}\left(x - \frac{1}{6}\pi\right)^3 - \frac{55}{384}e^{(1/2)}\left(x - \frac{1}{6}\pi\right)^4 + O\left(\left(x - \frac{1}{6}\pi\right)^5\right)$$

```
> f:=x->ln(sin(x))  ;taylor(f(x), x=Pi/3, 5);
```

$$f := x \to \ln(\sin(x))$$

$$\ln\left(\frac{1}{2}\sqrt{3}\right) + \frac{1}{3}\sqrt{3}\left(x - \frac{1}{3}\pi\right) - \frac{2}{3}\left(x - \frac{1}{3}\pi\right)^2 + \frac{4}{27}\sqrt{3}\left(x - \frac{1}{3}\pi\right)^3 - \frac{2}{9}\left(x - \frac{1}{3}\pi\right)^4 + \mathrm{O}\left(\left(x - \frac{1}{3}\pi\right)^5\right)$$

```
> f:=x->ln(1+x)/x^2 ;taylor(f(x), x=1, 4);
```

$$f := x \to \frac{\ln(1+x)}{x^2}$$

$$\ln(2) + \left(-2\ln(2) + \frac{1}{2}\right)(-1+x) + \left(3\ln(2) - \frac{9}{8}\right)(-1+x)^2 + \left(-4\ln(2) + \frac{43}{24}\right)(-1+x)^3 + \mathrm{O}((-1+x)^4)$$

```
> f:=x->sqrt(tan(x));taylor(f(x), x=Pi/4, 4);
```

$$f := x \to \sqrt{\tan(x)}$$

$$1 + x - \frac{1}{4}\pi + \frac{1}{2}\left(x - \frac{1}{4}\pi\right)^2 + \frac{5}{6}\left(x - \frac{1}{4}\pi\right)^3 + \mathrm{O}\left(\left(x - \frac{1}{4}\pi\right)^4\right)$$

```
> f:=x->sin(Pi*cos(x));taylor(f(x), x=Pi/3, 3);
```

$$f := x \to \sin(\pi\cos(x))$$

$$1 - \frac{3}{8}\pi^2\left(x - \frac{1}{3}\pi\right)^2 + \mathrm{O}\left(\left(x - \frac{1}{3}\pi\right)^3\right)$$

```
>
```

Surfaces - 3D

Groupe calcul formel - IREM de Grenoble

Fiche n° 22

```
> restart;
> plot3d(sin(x+sin(y)), x=-3..3, y=-3..3) ;
```



```
> plot3d([u*sin(t), u*cos(t), t/3], t=0..15, u=-1..1) ;
```



```
> with(plots) :
> tubeplot([10*cos(t), 10*sin(t), 0, t=0..2*Pi,
> radius=2+cos(7*t), numpoints=120, tubepoints=24],
> scaling= CONSTRAINED) ;
```



```
>
```

Annexe

Résumé des commandes utilisées

Pour toutes les commandes on peut utiliser l'aide en ligne de Maple V

Il suffit de taper sur une ligne vierge le caractère ? suivi du nom de la commande ou d'appuyer sur F1.
La page d'aide (en anglais) apparaît alors à l'écran. On trouve en fin de page des exemples d'utilisation.

Pour chacune des commandes décrites ici, nous présentons successivement :

- le nom en caractères gras
- une séquence d'appel avec la signification des paramètres
- une description de l'objet créé
- entre crochets, une fiche où la commande est utilisée

Description des commandes utilisées

@

@ est l'opérateur de composition. f@g, où f et g sont des fonctions, est la fonction fog.
[fonctions]

@@

@@ est l'opérateur de compositions réitérées.
Par exemple (f@@3) est la fonction fofof.
[fonctions]

abs

abs(x), où x est un nombre réel ou complexe, est la valeur absolue de x.
[nombres complexes]

argument **<unité polar>**

argument(x), où x est un nombre complexe, est un argument de x.
[nombres complexes]

array **<unité linalg>**

array(l), où l est une liste de listes, est la matrice ayant pour lignes les diverses listes constituant l.
[matrices]

binomial

binomial(n, p), où n et p sont des entiers, est le coefficient du binôme C_n^p.
[combinatoire]

charpoly **<unité linalg>**

charpoly(m, x), où m est une matrice carrée, x un nom de variable, est le polynôme caractéristique de la matrice m.
[matrices]

collect

collect(exp, x) est par exemple l'expression réduite et ordonnée d'un polynôme.
[polynômes]

combine

combine(exp) est l'expression transformée sous forme plus simple.
Il existe plusieurs options. Voir convert.
[intégration]

conjugate **<unité polar>**

conjugate(nc) est le conjugué du nombre complexe nc.
[complexes]

convert

convertit une expression sous différentes formes.
Il existe diverses options :

Changements de base :
convert(n, base2) est l'écriture en base base2 du nombre n écrit en base dix.
convert(n, **base**, base2) est la liste des chiffres en base base2 du nombre n écrit en base dix.
convert(l, **base**, base1, base2), où l est une liste de chiffres, base1, base2 des bases est la liste des chiffres en base2 de l écrit en base1.
convert(n, **binary**), où n est écrit en base dix, est le nombre binaire correspondant .
convert(expr, **confrac**) est une approximation de expr en fraction continue.
convert(nombre, **degrees**) conversion en degrés.
convert(expr, **expsincos**) conversion en expression avec des sinus et cosinus
convert(expr, **exp**) conversion en expression avec des exponentielles
convert(expr, **float**) conversion en expression flottante
convert(poly, **horner**) pour des polynômes
convert(expr, **ln**)
convert(objet, **list**) conversion en liste.
convert(nombre, **mod2**) conversion en nombre modulo 2.
convert(f, **parfrac**, x), où f est une fonction rationnelle, x le nom de la variable, est la décomposition de f en éléments simples.
convert(n, **polar**) est la forme polaire du complexe n.
convert(n, **radians**) conversion en radians
convert(expr, **radical**)
convert(expr, **sincos**) conversion en sinus et cosinus.
convert(expr, **trig**)
[nombres complexes]

D

D(f), où f est une fonction d'une variable, est la fonction dérivée.
(D@@2)(f) est la fonction dérivée seconde de f.
(D@@n)(f), où n est un entier connu, est la fonction dérivée d'ordre n de f.
D[i](f), où f est une fonction de plusieurs variables et i un entier, est la fonction dérivée partielle par rapport à la ième variable
D[i, j](f), où f est une fonction de plusieurs variables, i et j des entiers, est la fonction dérivée partielle seconde, par rapport aux variables de rang i et j.
[dérivation]

det < **unité linalg**>

det(m), où m est une matrice carrée est le déterminant de la matrice m.
[matrices]

diff

diff(f(x), x) est l'expression D(f)(x).
[fonctions]

dsolve

dsolve(eqn, f(x)), où eqn est une équation différentielle et f(x) est le nom de la variable, est l'ensemble des solutions de eqn.
Comme pour solve, il existe plusieurs options.
[équations différentielles]

eigenvals < **unité linalg**>

eigenvals(m), où m est une matrice carrée, est la séquence constituée des valeurs propres de la matrice m.
[matrices]

eigenvects < **unité linalg**>

eigenvects(m), où m est une matrice carrée, est une séquence constituée de listes. Chaque liste comprend une valeur propre de m, son ordre de multiplicité, et les vecteurs propres (listes) correspondants.
[matrices]

eval

eval(exp), où exp est une expression, est l'évaluation, la plus complète et exacte, de l'expression exp.

evalb

evalb(exp), où exp est une expression booléenne, est la valeur booléenne de exp.

evalc

evalc(n), où n est un nombre complexe (calculé), est l'évaluation du nombre n.
On obtient ainsi la partie réelle et la partie imaginaire de n
[nombres complexes]

evalf

evalf(exp), où exp est une expression numérique, est la valeur approchée de exp donnée avec un nombre fixé de chiffres significatifs. Ce nombre de chiffres est le contenu de la variable globale Digits.
evalf(exp, n) où exp est une expression numérique et n un entier positif, est la valeur approchée de exp donnée avec un n chiffres significatifs.
[calculs numériques]

evalm

evalm(exp), où exp est une expression matricielle, est l'évaluation exacte de exp.
[matrices]

expand

expand(exp), où exp est une expression algébrique, est l'expression développée.
[combinatoire]

factor

factor(exp), où exp est une expression algébrique, est l'expression factorisée.
[polynômes]

floor

floor(n), où n est un nombre réel, est la partie entière de n.
[calculs numériques]

fsolve

fsolve(eq), où eq est une équation, est l'ensemble des valeurs approchées des solutions de eq.
[calculs numériques]

gcd

gcd(p1, p2) est le pgcd des deux polynômes p1 et p2.
[polynômes]

gcdex

gcdex(p1, p2, x, 's', 't') voir igcdex.
[polynômes]

ifactor

ifactor(a), où a est un entier, est le produit des facteurs premiers de a.
a peut être aussi un rationnel.
[arithmétique]

igcd

igcd(a, b), où a et b sont des entiers, est le plus grand diviseur commun de a et b. [arithmétique]
igcd(x_1, x_2, ...x_n) est le plus grand diviseur commun des n entiers (n arbitraire).

igcdex

igcd(a, b, 'u', 'v'), où a et b sont des entiers, est le plus grand diviseur commun de a et b. 'u' et 'v' sont des noms. igcdex charge ces variables et leur donne pour contenu deux entiers tels que pgcd(a,b) = au +bv.
[arithmétique]

ilcm

ilcm(a,b), où a et b sont des entiers, est le plus petit multiple commun à a et b.
ilcm(x_1, x_2, ...x_n) est le plus petit multiple commun des n entiers (n arbitraire).
[arithmétique]

Im **< unité polar>**

Im(x), où x est un nombre complexe, est la partie imaginaire de x.
[nombres complexes]

int

int(f(x), x) est une primitive de f.
[fonctions]

integrate

integrate(f(x), x= a..b) est $\int_a^b f(x)dx$.
[fonctions]

iquo

iquo(D, d), où D et d sont des entiers, est le quotient de la division euclidienne de D par d. [arithmétique]

irem

irem(D, d), où D et d sont des entiers, est le reste de la division euclidienne de D par d. [arithmétique]

isprime

isprime(a), où a est un entier, est un booléen vrai si et ssi a est premier.
[arithmétique]

lcm

lcm(p1, p2) est le ppcm des deux polynômes p1 et p2.
[polynômes]

limit

limit(f(x), x=a) est la limite de f(x) quand x → a. a est un nombre ou bien **infinity**.
[fonctions]

plot

plot(f(x), x=xmin..xmax, y=ymin..ymax) crée, dans une fenêtre graphique, la courbe d'équation y = f(x). Les coordonnées extrêmes sont les valeurs xmin, xmax, ymin, ymax. La donnée de l'intervalle en y est optionnelle.
Pour représenter plusieurs fonctions la syntaxe est la suivante :
plot({f1(x), f2(x), ...}, x=xmin..xmax, y=ymin..ymax)
Pour représenter une ligne polygonale, le premier paramètre doit être une liste comportant, dans l'ordre, l'abscisse puis l'ordonnée de chaque point.
[fonctions]

plot3d

plot3d(f(x, y), x= x=xmin..xmax, y=ymin..ymax)) utilisée pour les graphiques en 3 dimensions comporte de multiples options.
[surfaces]

polar < **unité polar**>

polar(x), où x est un nombre complexe, est la forme polaire de x.
[nombres complexes]

product

product(u(p), p=pmin..pmax) est le produit des u(p)lorsque p décrit l'intervalle pmin..pmax.
[nombres complexes]

Re < **unité polar**>

Re(x), où x est un nombre complexe, est la partie réelle de x.
[nombres complexes]

readlib

readlib(fich.m) permet d'inclure les commandes décrites et compilées dans le fichier fich.m.
[nombres complexes]

seq

seq(f(i), i = m..n), où f est une fonction, i un nom de variable, m et n des valeurs numériques, est la séquence des valeurs de f(i) lorsque i décrit l'intervalle m..n.
[arithmétique]

select

select(x->B(x), E), où B est une fonction booléenne, et E un ensemble.
[probabilités]

simplify

simplify(expr), où expr est une expression, est l'expression simplifiée.

Il existe diverses options :

simplify(expr, **radical**) pour des exposants rationnels
simplify(expr, **sqrt**) pour des racines carrées
simplify(expr, **power**) pour des puissances
simplify(expr, **trig**) en trigonométrie
simplify(expr, **exp**) pour des exponentielles
simplify(expr, **ln**) pour ln

[nombres complexes]

solve

solve(eqns, nv), où eqns est une équation ou un système et nv un nom de variable ou l'ensemble des noms des variables, est l'ensemble des solutions exactes de eqns.

Il existe diverses champs d'application :

solutions approchées (cf fsolve)
équations fonctionnelles
inéquations
systèmes linéaires (voir linsolve dans la bibliothèque linalg)
équations irrationnelles

[nombres complexes]

sqrt

sqrt(x), où x est un nombre réel, est le radical de x.

[calculs numériques]

subs

subs(x=a, f(x)), où a est un objet, est l'expression f(a).

[matrices]

sum

sum(f(i), i = m..n), où f est une fonction, i un nom de variable, m et n des valeurs numériques, est la somme des valeurs de f(i) lorsque i décrit l'intervalle m..n.

[combinatoire]

taylor

taylor(f(x), x=a, o), où o est un nombre entier, est le développement de Taylor de f(x) au voisinage de a, à l'ordre o.

[développements limités]

time()

time() donne le temps, en secondes, écoulé depuis le début de la session.

[combinatoire]

unapply

unapply(f(x), x) est la fonction : x → f(x).

[fonctions]

with

with(nom), où nom est un nom de bibliothèque Maple V, rend accessible les commandes définies dans cette bibliothèque.

[matrices]